

บรรณานุกรม

1. Bassous, E., "Fabrication of Novel Three Dimentional Microstructures by The Anistropic Etching of (100) and (110) Silicon, "IEEE Transaction on Electron Devices, 25 (10), pp. 1178-1185, 1978.
2. Rebeiz, G., D. Rutledge, "Imaging Antenna Structure for Submillimeter Wavelengths," NASA Tech Briefs, 8 (14), pp. 24, 1990.
3. McCauhan, D.V., J.C.White, "MOS Transistor and Memories," Handbook on Semiconductors (Moss, T.S.), pp. 299, North-Holland, Netherlands, 1981.
4. Roger, T.H., R.S. Muller, "Silicon Micromechanics: Sensor and Actuators on a Chip," IEEE Spectrum, 7 (27), pp. 30, 1990.
5. Kenneth, E.B., "Anisotropic Etching of Silicon," IEEE Transaction on Electron Devices, 25 (10), pp. 1185-1193, 1978.
6. มนตรี สวัสดิ์ศฤงฆาร, สมศักดิ์ ปัญญาแก้ว, บรรยง โตประเสริฐพงศ์ และ เกียรติศักดิ์ เจริมดีระกุล, สมบัติของวัสดุทางวิศวกรรมไฟฟ้า, หน้า 26-33, ภาควิชาวิศวกรรมไฟฟ้า คณะวิศวกรรมศาสตร์ จุฬาลงกรณ์มหาวิทยาลัย, กรุงเทพมหานคร.
7. สุทิน เวทธีวัฒน์, "เซลล์แสงอาทิตย์แบบหัวต่อ พีเอ็น," รายงานโครงการวิจัย, คณะวิศวกรรมศาสตร์ จุฬาลงกรณ์มหาวิทยาลัย, กรุงเทพมหานคร, 2524.
8. Stevens, H.P., "Making Ultrathin Solar Cells," Technical Support Package, Jet Propulsion Laboratory, California, 1991.
9. Bube, R.H., "Solar Cells," Handbook on Semiconductors (Mass, T.S.), pp. 691-714, North-Hollands, Netherlands, 1981.

10. สมศักดิ์ ปัญญาแก้ว, "การศึกษาออกแบบระบบโฟโตโวลตาอิก," รายงานโครงการวิจัย, คณะวิศวกรรมศาสตร์ จุฬาลงกรณ์มหาวิทยาลัย, กรุงเทพมหานคร, 2527.
11. สมศักดิ์ ปัญญาแก้ว, เทคโนโลยีเซลล์แสงอาทิตย์, หน้า 86-150, จุฬาลงกรณ์มหาวิทยาลัย, กรุงเทพมหานคร, 2530.
12. ธนวิษฐ์ ชูลิภาวิทย์, "การศึกษากระบวนการผลิตตัวเก็บประจุมอสและทรานซิสเตอร์แบบมอส," วิทยานิพนธ์ปริญญามหาบัณฑิต, ภาควิชาวิศวกรรมไฟฟ้า บัณฑิตวิทยาลัย จุฬาลงกรณ์มหาวิทยาลัย, 2532.
13. Dieter, K.S., Semiconductor Material and Device Characterization, pp. 147-160, Wiley, New York, 1990.
14. ชุมพล อันตรเสน, "การประดิษฐ์เซลล์แสงอาทิตย์แบบหัวต่อ พีเอ็น จากแวนดิลิกซิลิคอนซึ่งมีสภาพผิวขรุขระต่าง ๆ กัน," วิทยานิพนธ์ปริญญามหาบัณฑิต, ภาควิชาวิศวกรรมไฟฟ้า บัณฑิตวิทยาลัย จุฬาลงกรณ์มหาวิทยาลัย, 2523.
15. บรรยง โตประเสริฐพงศ์, มนต์วี สวัสดิ์ศฤงฆาร, เกียรติศักดิ์ เจริมดิระกุล และ สมศักดิ์ ปัญญาแก้ว, "การปรับปรุงกระบวนการผลิตเพื่อเพิ่มประสิทธิภาพของเซลล์แสงอาทิตย์ที่ทำจากซิลิคอนแบบโพลีคริสตัลไลน์," รายงานโครงการวิจัย, เลขที่ 26 G-EE-2529, คณะวิศวกรรมศาสตร์ จุฬาลงกรณ์มหาวิทยาลัย, กรุงเทพมหานคร, 2531.
16. Martin, A.G., Solar Cells, pp. 4-16, Prentice-Hall, New Jersey, 1982.

ภาคผนวก

ภาคผนวก ก.

โปรแกรมการวัดค่าการสะท้อนแสง

```

100 '----- OPTICAL POWER MEASUREMENT -----'
110 '                for Anrisu ML93B                '
120 '                -----'
130 '                19/07/1991 by Suwat Sopitpan        '
140 '-----'
150 '
160 '
180 CONSOLE 0,25,0,1 : CLS : OPTION BASE 1
185 DELAY = 9000 : SELECT = 1 : DOG = 4
190 DIM A$(18)
200 DIM A(200), C(200)
205 DIM WLENGTH(200) , CONTENT(200)
210 *SHOW
215 CONSOLE 0,25,0,1 : CLS
220 IF ( SELECT = DOG ) GOTO 990
225 GOSUB *MENU
230 *LOOP1
240 LOCATE 15,20 : COLOR 7 : INPUT "Select : ",SELECT
250 ON SELECT GOSUB *OPTICAL, *FSAVE, *EPSON, *BYE
260 GOTO *SHOW
999 '
1000 *DATSET '----- DATA SET -----'
1010 WLI = 400 'initial wavelength > 400 nm <
1020 WLF = 800 'final wavelength > 800 nm <
1030 WLS = 10 'step > 10 nm <
1040 RETURN
1050 '
1070 *DEVZERO '----- DEVICE ZERO SETTING -----'
1073 GOSUB *DEVSET
1075 LOCATE 30,20 : COLOR 5
1077 WRITE "Please wait" : FOR I = 1 TO 999 : NEXT I
1078 'LOCATE 30,20 : COLOR 4
1079 'WRITE "Please wait" : FOR I = 1 TO 999 : NEXT I
1080 PRINT @15; "Z" 'zero setting
1090 INPUT @15; A$ 'wait for zero setpoint
1100 LOCATE 30,20 : COLOR 2
1110 WRITE "Zero set ready" : FOR I = 1 TO 4000 : NEXT I
1120 GOSUB *DEVRESET
1130 '
1140 RETURN
1150 '
1160 *DEVSET '----- DEVICE SET IEEE CONTROLLER -----'
1170 ISET IFC
1180 ISET REN
1190 CMD DELIM = 0
1200 RETURN

```



```

1210 '
1220 *DEVRESET '----- RESET TO LOCAL CONTROL -----'
1230 IRESET REN
1240 RETURN
1250 '
1260 '
1270 '
1280 *MEASURE '----- measure -----'
1285 '
1290 CLS : COLOR 4 : LOCATE 10,9
1300 PRINT "-- Optical measurement --" : COLOR 5
1310 LOCATE 10,11 : INPUT "Start wavelength : ",INIT
1320 LOCATE 10,13 : INPUT "Stop wavelength : ",B
1330 GOSUB *DEVSET : GOSUB *SETPARA
1335 LOCATE 10,15 : COLOR 6 : PRINT "WAVELENGTH          POWER"
1337 N = 1 : A(N) = INIT
1340 *LOOP2
1350 INPUT "", Z$
1352 INPUT @15; A$ : GOSUB *TOOD
1355 B$ = MID$(A$,2,1) : C(N) = (ASC(B$)-79)*(-3)
1357 B$ = "1E" + STR$(C(N)) : C(N) = VAL(MID$(A$,7,6)) * VAL(B$)
1360 LOCATE 12,17 : COLOR 2 : PRINT A(N), C(N)
1370 N = N + 1 : A(N) = A(N-1) + 10
1380 IF A(N) > B THEN *EXIT2
1385 ' FOR I = 1 TO DELAY : NEXT I
1390 GOTO *LOOP2
1400 *EXIT2
1410 BEEP 1 : FOR C = 1 TO DELAY : NEXT C : BEEP 0
1420 '
1430 MAXOPT = N-1
1440 GOSUB *DEVRESET
1450 RETURN
1460 '
1500 *SETPARA '----- measure parameters -----'
1510 PRINT @15; "D+0000" 'CAL.FACTOR
1520 PRINT @15; "C0" 'WATT
1530 PRINT @15; "F1" 'FREE RUN
1540 PRINT @15; "B" 'AUTO RANGE
1550 '
1600 '
1610 RETURN
1620 '
2000 *OPTICAL
2010 CONSOLE 6,24,0,1 : CLS
2020 COLOR 4
2025 LOCATE 20,6 : PRINT "...OPTICAL..."
2027 COLOR 6
2030 LOCATE 20,8 : PRINT "1. Zero power set"
2040 LOCATE 20,9 : PRINT "2. Start measurement"
2050 LOCATE 20,10: PRINT "3. Quit to main"
2060 '
2070 '
2080 '
2090 '
2100 LOCATE 20,14 : COLOR 7
2110 INPUT "Select : ",SELECT

```

```

2120 ON SELECT GOSUB *DEVZERO, *MEASURE
2130 IF SELECT = 3 THEN *OPTICEND
2140 GOTO *OPTICAL
2150 *OPTICEND
2160 RETURN
2170 '
4400 *TOOD
4410 BEEP 1
4420 FOR TD = 1 TO 500 : NEXT TD
4430 BEEP 0
4440 RETURN
6050 '
7000 *FSAVE
7010 COLOR 3 : CLS : LOCATE 20,9
7020 PRINT "---- Save data to disk ----"
7030 COLOR 5 : LOCATE 10,11
7050 LOCATE 20,12 : INPUT "File name : ",F$
7060 LOCATE 20,13 : INPUT "Comment : ",COMMT$
7065 *SAVEASK
7070 LOCATE 20,16 : INPUT "Sure (1), Not correct (2) or Quit (3) : ",ASK
7080 ON ASK GOTO *GOSAVE, *FSAVE, *EXITSAVE
7090 GOTO *SAVEASK
7100 *GOSAVE
7140 EXTEN$ = ".opt"
7150 FOR I = 1 TO MAXOPT
7160 WLENGTH(I) = A(I) : CONTENT(I) = C(I)
7170 NEXT I
7300 *SAVEFILE
7310 FILE$ = F$ + EXTEN$
7320 OPEN FILE$ AS #1
7330 WRITE #1 , COMMT$, I-1
7340 FOR N = 1 TO I-1
7350 WRITE #1 , WLENGTH(N), CONTENT(N)
7360 NEXT N
7370 CLOSE #1
7380 *EXITSAVE
7390 RETURN
8050 '
8500 *EPSON
8510 LPRINT "-----"
8512 LPRINT "---- Optical power MA97B ----"
8520 LPRINT " Wavelength [nm]   Power [watt] "
8530 LPRINT "-----"
8540 FOR N = 1 TO MAXOPT
8550 LPRINT A(N), C(N)
8560 NEXT N
8570 RETURN
8575 '
9000 *BYE
9010 SELECT = DOG : CLS
9020 RETURN
9060 '
9280 *MENU : CLS
9290 COLOR 5 : LOCATE 10,3
9300 PRINT "----- OPTICAL POWER MEASUREMENT -----"
9310 COLOR 2 : LOCATE 10,5

```



```
9320 PRINT "-----"  
9330 LOCATE 20,8 : COLOR 3  
9360 PRINT "1. Start measurement"  
9370 LOCATE 20,10  
9420 PRINT "2. Store data to disk"  
9430 LOCATE 20,12  
9460 PRINT "3. Print data to printer"  
9465 LOCATE 20,14  
9467 PRINT "4. Quit"  
9470 RETURN
```

ภาคผนวก ข.

โปรแกรมการวัดลักษณะสมบัติของเซลล์แสงอาทิตย์ในสภาวะมืด

```

1000      '*****
1010      '*
1020      '*      CURRENT(J)-VOLTAGE(V) CHARACTERISTICS (dark)      '*
1030      '*
1040      '*      BY 4140B pA METER / DC VOTAGE SOURCE      '*
1050      '*      (SUBDIRECTLY STRUCTURE FILE VERSION)      '*
1060      '*****
1070      '
1080      '      Latestly edited in 91/07/10 by SUWAT SOPITPAN
1090      '
1150      '----- MAIN PROGRAM -----
1160      CLEAR , , , &H6D6
1170      GOSUB *DATASET
1180      *MAIN
1190      CLS 3:CONSOLE 0,25,0,1 :SCREEN 3,1
1200      COLOR 6 :LOCATE 19,3 :PRINT "*****I-V CHARACTERISTIC*****"
1210      COLOR 5:LOCATE 28,5 :PRINT "TASK SELECTION"
1220      LOCATE 24,7 :PRINT "  AUTO MEASUREMENT"
1230      LOCATE 24,8 :PRINT "  DATA TREATMENT"
1240      COLOR 2:LOCATE 24,10:PRINT"PRESS YOUR CHOICE NUMBER":COLOR 5
1250      COLOR@ (24,7)-(25,8),2
1260      COLOR 4 :LOCATE 18,15 :PRINT "<Function key information>"
1270      LOCATE 17,17 :PRINT "[f.8]--Change directly"
1280      LOCATE 17,18 :PRINT "[f.10]--Make new directly"
1290      *LOOP1:C$=INKEY$
1300      IF C$="" THEN *LOOP1
1310      IF C$="1" THEN GOSUB *MEASURE
1320      IF C$="2" THEN GOSUB *DATATRE
1330      GOTO *MAIN
1340      END
1350      '
1390      '----- DATA SET -----
1400      *DATASET :C1=LOG(10)
1410      DIM IPXLAB(50),IPYLAB(50),XLAB(50),YLAB(50),SIG(1000)
1420      DIM PS(1000),TS(1000),X2(12),X4(12),P(1000),T(1000)
1430      DIM XLAB$(50),YLAB$(50),A1$(500),BB$(500),IX(1000),IY(1000)
1440      ON KEY GOSUB *K1,*K2,*K3,*K4,*K5,*K6,*K7,*DIRCH,*K9,*DIRMK
1450      KEY(8) ON :KEY(10) ON
1460      FOR M=2 TO 12 :X2(M)=2*(M*M*M/3+M*M/2+6) :NEXT M
1470      FOR M=2 TO 12 :X4(M)=2*(M*M*M*M*M/5+M*M*M*M/2+M*M*M/3-M/30)
1475      NEXT M
1480      PS=0 :PT=1 :PE= .01 :BOS= 1 :PD= .1 :AREA= 1
1490      DV$="A:"
1500      YTIT$="Current (Amp)": XTIT$="Voltage (V)"
1510      RETURN
1520      '

```



```

1530 ' ----- SMOOTH or NOT -----
1540 *SMOOTH
1550 CLS 1 :COLOR 5
1560 *SM1:LOCATE 21,5 :INPUT "Smoothing Points (5<=odd<=25)";NS
1570 IF (NS MOD 2=0) OR (NS<5) OR (NS>25) THEN *SM1
1580 FOR I=1 TO ITOT :TS(I)=T(I) :NEXT I
1590 M=(NS-1)/2
1600 FOR I=1+M TO ITOT-M
1610 SUM0=0 :SUM1=0
1620 FOR J=-M TO M
1630 SUM0=SUM0+P(I+J)*(X4(M)-X2(M)*J*J)
1640 SUM1=SUM1+P(I+J)*J
1650 NEXT J
1660 'PS(I)=(SUM1/X2(M))/(SUM0/(NS*X4(M)-X2(M)*X2(M)))
1670 PS(I)=SUM1/X2(M)
1680 NEXT I
1690 IE=1
1700 FOR I=1+M TO ITOT-M
1710 P(IE)=PS(I)
1720 T(IE)=TS(I)
1730 IE=IE+1
1740 NEXT I
1750 ITOT=ITOT-2*M
1760 RETURN
1770 '
1780 '----- DATA TREATMENT -----
1790 *DATATRE:CLS 3
1800 COLOR 6:LOCATE 27,0
1810 PRINT "*** DATA TREATMENT ***"
1820 COLOR 5:LOCATE 31,2:PRINT "MODE SELECTION"
1830 LOCATE 27,4 :PRINT "1 DATA DISPLY"
1840 LOCATE 27,6 :PRINT "2 FILE MANAGEMENT"
1850 LOCATE 27,8 :PRINT "3 GO TO TASK SELECTION"
1860 LOCATE 27,10:PRINT "PRESS YOUR CHOICE NUMBER"
1870 COLOR@ (27,4)-(27,10),2:COLOR@ (27,10)-(51,10),2
1880 *LOOP100:C$=INKEY$
1890 IF C$="" THEN *LOOP100
1900 IF C$="1" THEN GOSUB *DIRCH :*DATADISP
1910 IF C$="2" THEN *FILEMANA
1920 IF C$="3" THEN RETURN
1930 GOTO *DATATRE
1940 '
1950 '----- DATA DISPLY -----
1960 *DATADISP
1970 CONSOLE 0,25,0,1 :CLS 1 :COLOR 7
1980 COLOR 5: ' GOSUB *FILEREF
1985 LINE INPUT "FILE NAME : ",FILE$
1990 GOSUB *MYLOAD
2000 FC$=FILE$+"//"+COMMEN$
2010 IF IRE=1 THEN *MYLOAD
2020 GOSUB *LDDISP
2030 'GOSUB *TUNNEL
2040 *DISPLY
2050 FOR I=1 TO ITOT
2060 P(I)=ABS(P(I)) :T(I)=ABS(T(I))
2070 NEXT I

```

```

2090 GOSUB *MINMAX
2092 GOSUB *XYDTMN
2094 GOSUB *COMGRA
2100 GOSUB *CRT
2102 'GOSUB *SDATA
2103 GOSUB *HC
2105 '
2108 GOTO *DISPLY
2110 '
2120 '----- MINMUM AND MAXMUN -----
2130 *MINMAX :PMAX=-1E+30 :PMIN=1E+30 :TMAX=-1E+30 :TMIN=1E+30
2140 FOR I=1 TO ITOT
2150 IF P(I)<PMIN THEN PMIN=P(I)
2160 IF P(I)>PMAX THEN PMAX=P(I)
2170 IF T(I)<TMIN THEN TMIN=T(I)
2180 IF T(I)>TMAX THEN TMAX=T(I)
2190 NEXT I
2200 'TMIN=VSTA :TMAX=VSTO
2210 'IF BOS=0 THEN TMIN=-VSTO
2220 RETURN
2230 '
2240 '
2250 '----- XMIN,XMAX,YMIN,YMAX DETERMINATION -----
2260 *XYDTMN
2270 CONSOLE 0,25,0,1 :CLS 1 :COLOR 7
2280 PRINT "XMIN=";TMIN
2290 PRINT "XMAX=";TMAX
2300 PRINT "YMIN=";PMIN
2310 PRINT "YMAX=";PMAX :COLOR 6
2320 LOCATE 16,0 :PRINT "PLOTTING TASK and MODE SELECTION" :COLOR 5
2330 LOCATE 21,2 :PRINT "ID X-axis - Y-axis" :COLOR 0
2340 LOCATE 21,3 :PRINT "0 LINEAR - LINEAR"
2350 LOCATE 21,4 :PRINT "1 LOG - LINEAR"
2360 LOCATE 21,5 :PRINT "2 LINEAR - LOG"
2370 LOCATE 21,6 :PRINT "3 LOG - LOG"
2380 LOCATE 21,8 :PRINT "4 New Ref. Loading "
2390 LOCATE 21,9 :PRINT "5 GOTO STOREDATA "
2400 LOCATE 21,10:PRINT "6 NEXT MEASUREMENT"
2405 LOCATE 21,11:PRINT "7 FORMFEED "
2410 COLOR@ (21,3)-(21,11),2 :COLOR@ (25,3)-(45,11),5 :COLOR 5
2420 LOCATE 11,12 :INPUT "Type ID # to choose ITEM ";IOP0
2430 IF (IOP0<0) OR (IOP0>7) THEN *XYDTMN
2440 IF IOP0=4 THEN RETURN *DATADISP
2450 IF IOP0=5 THEN GOSUB *STOREDATA :RETURN *DISPLY
2460 IF IOP0=6 THEN RETURN *MEASURE
2465 IF IOP0=7 THEN LPRINT CHR$(12)
2467 IF IOP0>3 THEN *XYDTMN
2470 IOP11=IOP0 MOD 2
2480 IOP12=IOP0 ¥ 2
2490 *JUMP11
2500 LOCATE 21,14 :PRINT "X-Axis From ( ? )";TAB(41);TMIN
2510 LOCATE 21,15 :PRINT " To ( ? )";TAB(41);TMAX
2520 LOCATE 21,16 :PRINT "Y-Axis From ( ? )";TAB(41);PMIN
2530 LOCATE 21,17 :PRINT " To ( ? )";TAB(41);PMAX
2540 COLOR@ (21,14)-(55,14),7 :LOCATE 55,14 :PRINT SPC(20)
2550 LOCATE 55,14 :INPUT "";XMIN$

```



```

2560 IF XMIN$="" THEN XMIN=TMIN ELSE XMIN=VAL(XMIN$)
2570 LOCATE 39,15 :PRINT SPC(1) :COLOR@ (21,14)-(79,14),5
2580 COLOR@ (21,15)-(55,15),7 :LOCATE 55,15 :PRINT SPC(20)
2590 LOCATE 55,15 :INPUT "";XMAX$
2600 IF XMAX$="" THEN XMAX=TMAX ELSE XMAX=VAL(XMAX$)
2610 LOCATE 39,15 :PRINT SPC(1) :COLOR@ (21,15)-(79,15),5
2620 IF XMIN>=XMAX THEN *JUMP11
2630 IF (IOP11=1) AND ((XMIN<=0) OR (XMAX<=0)) THEN *JUMP11
2640 *JUMP12 :COLOR@ (21,16)-(55,16),7 :LOCATE 55,16 :PRINT SPC(20)
2650 LOCATE 55,16 :INPUT "";YMIN$
2660 IF YMIN$="" THEN YMIN=PMIN ELSE YMIN=VAL(YMIN$)
2670 LOCATE 39,16 :PRINT SPC(1) :COLOR@ (21,16)-(79,16),5
2680 COLOR@ (21,17)-(55,17),7 :LOCATE 55,17 :PRINT SPC(20)
2690 LOCATE 55,17 :INPUT "";YMAX$
2700 IF YMAX$="" THEN YMAX=PMAX ELSE YMAX=VAL(YMAX$)
2710 LOCATE 39,17 :PRINT SPC(1) :COLOR@ (21,17)-(79,17),5
2720 IF YMIN>=YMAX THEN *JUMP12
2730 IF (IOP12=1) AND ((YMIN<=0) OR (YMAX<=0)) THEN *JUMP12
2740 LOCATE 15,22 :COLOR 4
2750 PRINT "Crear this figure?(1) or not(2) "
2760 *LOOP302 :A$=INKEY$
2770 IF A$="" THEN *LOOP302
2780 CORN=VAL(A$) :CNUM=3
2790 IF CORN=2 THEN CNUM=1 ELSE CNUM=3
2800 RETURN
2810 '
2820 ' ----- HARD COPY THE GRAPH ON CRT or NOT -----
2830 *HC
2840 LOCATE 50,3 :PRINT "***** Hard Copy *****"
2850 LOCATE 50,4 :PRINT "DATA -----(1)"
2860 LOCATE 50,5 :PRINT "Figure and Inform. -----(2)"
2870 LOCATE 50,6 :PRINT "PLOTTER -----(3)"
2880 LOCATE 50,7 :PRINT "DATA to RS232 ----(4)"
2882 LOCATE 50,8 :PRINT "INFO & DATA to RS232 ----(5)"
2885 LOCATE 50,9 :PRINT "Not -----(RETURN)"
2890 *LOOP300 :A$=INKEY$: IF A$="" THEN *LOOP300
2900 IF A$="1" OR A$="2" THEN LPRINT SAMPLE$;SPC(5);COMMEN$
;SPC(15);DATTE$ :LPRINT FC$

2910 IF A$="1" THEN GOSUB *PRDATA
2920 IF A$="2" THEN LPRINT CHR$(27);"G" :CHILD "HCOPI"
:LPRINT CHR$(27);"H"

2930 'IF A$="1" OR A$="2" THEN LPRINT CHR$(12)
2940 IF A$="3" THEN GOSUB *PLOTTER :RETURN
2950 IF A$="4" OR A$="5" THEN OPEN "com:n81xn" AS #2 ELSE RETURN
2960 IF A$="5" THEN GOSUB *INFOTRAN
2965 GOSUB *DATATRAN
2970 CLOSE #2
2975 RETURN
2980 ' ----- COMMON GRAPHIC ROUTINE ON CRT AND PLOTTER -----
2990 *COMGRA
3000 ZMAX=XMAX
3010 ZMIN=XMIN
3020 IOP1=IOP11
3030 GOSUB *SCALE
3040 IOPX3=IOP3
3050 IOPX4=IOP4

```

```

3060 NXT=NZT
3070 XORG2=ZORG2
3080 XT=ZT
3090 ZMAX=YMAX
3100 ZMIN=YMIN
3110 IOP1=IOP12
3120 GOSUB *SCALE
3130 IOPY3=IOP3
3140 IOPY4=IOP4
3150 NYT=NZT
3160 YORG2=ZORG2
3170 YT=ZT
3180 RETURN
3190 '
3200 ' ----- SCALING SUBROUTINE -----
3210 *SCALE
3220 IF IOP1=1 THEN *JUMP13
3230 LZ=INT(LOG(ZMAX)/C1)
3240 ZT=10^LZ
3250 IZORG1=INT(ZMIN/ZT)
3260 ZORG1=ZT*IZORG1
3270 IZORG2=INT(ZMAX/ZT)
3280 ZORG2=ZT*IZORG2
3290 IF CINT(IZORG2)<>CINT(ZMAX/ZT) THEN IZORG2 = IZORG2+1
                                     :ZORG2 = ZORG2+ZT

3300 GOTO *JUMP14
3310 *JUMP13:OZ=LOG(ZMAX)/C1
3320 IZORG2=INT(OZ)
3330 IF IZORG2<>OZ THEN IZORG2=IZORG2+1
3340 IZORG1=INT(LOG(ZMIN)/C1)
3350 ZORG2=IZORG2
3360 ZORG1=IZORG1
3370 ZT=1
3380 *JUMP14
3390 IOP2=1
3400 NZT=IZORG2-IZORG1
3410 IF (IOP1=0) AND (NZT=1) THEN ZT=ZT*.1:NZT=10
3420 IF (IOP1=0) AND (NZT=2) THEN IOP2=2
3430 NZT=NZT*IOP2
3440 ZT=ZT/IOP2
3450 IOP3=1
3460 IF (IOP1=0) AND ((NZT>5) OR (IOP2=2)) THEN IOP3=2
3470 IOP4=1
3480 IF (IOP1=0) AND (IZORG2 MOD 2<>0) AND
    (IZORG2-IZORG1<>1) AND (IOP3=2) AND (IOP2=1) THEN IOP4=2
3490 RETURN
3500 '
3510 ' ----- PLOT ON CRT -----
3520 *CRT
3530 IPXORG1=116
3540 IPXORG2=388
3550 IPYORG1=340
3560 IPYORG2=10
3570 CONSOLE 0,25,0,1 :SCREEN 3,0,0,0 :COLOR 7,0,0,5
3580 IPXTIT=(IPXORG1+IPXORG2)*.5-LEN(XTIT$)*7
3590 IPYTIT=(IPYORG1+IPYORG2)*.5-LEN(YTIT$)*10

```



```

3600 PXT=(IPXORG2-IPXORG1)/NXT
3610 FOR I=1 TO NXT+1
3620 IPXLAB(I)=IPXORG2-PXT*(I-1)
3630 XLAB$(I)=STR$(XORG2-XT*(I-1))
3640 'IF IOP11=1 THEN XLAB$(I)="1E"+XLAB$(I)
3650 NEXT I
3660 PYT=(IPYORG1-IPYORG2)/NYT
3670 FOR I=1 TO NYT+1
3680 IPYLAB(I)=IPYORG2+PYT*(I-1)
3690 YY=YORG2-YT*(I-1)
3700 IF ABS(YY)<1E-12 THEN YY=0
3710 YLAB$(I)=STR$(YY)
3720 NEXT I
3730 A=PXT/XT:B=PYT/YT
3740 FOR I=1 TO ITOT
3750 IF IOP11=1 THEN *JUMP15
3760 IX(I)=IPXORG2+(T(I)-XORG2)*A :GOTO *JUMP16
3770 *JUMP15
      :IF T(I)<=0 THEN IX(I)=IPXORG1 :GOTO *JUMP16
3780 IX(I)=IPXORG2+(LOG(T(I))/C1-XORG2)*A
3790 *JUMP16
      :IF (IX(I)<IPXORG1) OR (IX(I)>IPXORG2) THEN IX(I)=IPXORG1
3800 IF IOP12=1 THEN *JUMP17
3810 IY(I)=IPYORG2-(P(I)-YORG2)*B :GOTO *JUMP18
3820 *JUMP17
      :IF P(I)<=0 THEN IY(I)=IPYORG1 :GOTO *JUMP18
3830 IY(I)=IPYORG2-(LOG(P(I))/C1-YORG2)*B
3840 *JUMP18
      :IF (IY(I)<IPYORG2) OR (IY(I)>IPYORG1) THEN IY(I)=IPYORG1
3850 NEXT I :AAA=A :BBB=B
3860 CLS CNUM :SCREEN 3,0,0,1
3870 POINT (IPXORG2,IPYORG1)
3880 FOR I=2 TO NXT+1
3890 LINE -(IPXLAB(I),IPYORG1) :LINE -STEP(0,-8) :POINT STEP(0,8)
3900 NEXT I
3910 FOR I=NYT TO 1 STEP -1
3920 LINE -(IPXORG1,IPYLAB(I)) :LINE -STEP(4,0) :POINT STEP(-4,0)
3930 NEXT I
3940 FOR I=NXT TO 1 STEP -1
3950 LINE -(IPXLAB(I),IPYORG2) :LINE -STEP(0,8) :POINT STEP(0,-8)
3960 NEXT I
3970 FOR I=2 TO NYT+1
3980 LINE -(IPXORG2,IPYLAB(I)) :LINE -STEP(-4,0) :POINT STEP(4,0)
3990 NEXT I
4000 FOR I=IOPX4 TO NXT+1 STEP IOPX3
4010 A=LEN(XLAB$(I))
4020 IF IOP11=1 THEN *JUMP19
4030 PXL=8*A*.5
4040 FOR J=1 TO A
4050 A2$=MID$(XLAB$(I),J,1)
4060 PUT (IPXLAB(I)-PXL+8*(J-1),IPYORG1+6),KANJI(ASC(A2$)),PSET
4070 NEXT J
4080 GOTO *JUMP20
4090 *JUMP19 :PUT(IPXLAB(I)-15,IPYORG1+13),KANJI(49),PSET
4100 PUT (IPXLAB(I)-7,IPYORG1+13),KANJI(48),PSET
4110 FOR J=1 TO A

```

```

4120 A2$=MID$(XLAB$(I),J,1)
4130 PUT (IPXLAB(I)+7*(J-1),IPYORG1+5),KANJI(ASC(A2$)+256),PSET
4140 NEXT J
4150 *JUMP20 :NEXT I
4160 AA=0
4170 FOR I=IOPY4 TO NYT+1 STEP IOPY3
4180 A=LEN(YLAB$(I))
4190 IF A>AA THEN AA=A
4200 NEXT I
4210 FOR I=IOPY4 TO NYT+1 STEP IOPY3
4220 A=LEN(YLAB$(I))
4230 IF IOP12=1 THEN *JUMP21
4240 PYL=8*A+5
4250 FOR J=1 TO A
4260 A2$=MID$(YLAB$(I),J,1)
4270 PUT (IPXORG1-PYL+8*(J-1),IPYLAB(I)-7),KANJI(ASC(A2$)),PSET
4280 NEXT J
4290 GOTO *JUMP22
4300 *JUMP21 :PUT (IPXORG1-8*AA-15,IPYLAB(I)),KANJI(49),PSET
4310 PUT (IPXORG1-8*AA-7,IPYLAB(I)),KANJI(48),PSET
4320 FOR J=1 TO A
4330 A2$=MID$(YLAB$(I),J,1)
4340 PUT (IPXORG1+8*(-AA+J-1),IPYLAB(I)-8),KANJI(ASC(A2$)+256),PSET
4350 NEXT J
4360 *JUMP22 :NEXT I
4370 A=LEN(XTIT$)
4380 B=IPYORG1
4390 IF IOP11=1 THEN B=B+38 ELSE B=B+30
4400 FOR I=1 TO A
4410 A2$=MID$(XTIT$,I,1)
4420 PUT (IPXTIT+14*(I-1),B),KANJI(ASC(A2$)),PSET,5,0
4430 NEXT I
4440 A=LEN(YTIT$)
4450 B=IPXORG1-10-AA*8
4460 IF IOP12=1 THEN B=B-28
4470 FOR I=1 TO A
4480 A2$=MID$(YTIT$,I,1)
4490 PUT (B,IPYTIT+17*(I-1)),KANJI(ASC(A2$)),PSET,5,0
4500 NEXT I
4510 FOR I=1 TO ITOT
4520 HANKEI = 1 'IF I>ITOT/2 THEN HANKEI=2 ELSE HANKEI=3
4530 IF I>ITOT/2 THEN CCOL=1 ELSE CCOL=2
4540 CIRCLE(IX(I),IY(I)),HANKEI,CCOL
4550 NEXT I
4560 LINE (IPXORG1,IPYORG2+BBB*YORG2)-(IPXORG2,IPYORG2+BBB*YORG2),5
4570 LINE (IPXORG2-AAA*XORG2,IPYORG1)-(IPXORG2-AAA*XORG2,IPYORG2),5
4580 RETURN
4590 '
4600 ' ----- PLOT ON PLOTTER -----
4610 *PLOTTER
4620 ISET IFC :ISET REN
4630 IPXMAX=3000 :IPYMAX=2750
4640 IPXORG1=350 :IPXORG2=IPXORG1+1000
      :IPYORG1=420 :IPYORG2=IPYORG1+1250
4650 IPXDATE=IPXORG1-150 :IPYDATE=IPYORG2+450
      :IPXSAM=IPXORG1-150 :IPYSAM=IPYORG2+350

```



```

4660 IPXFAC=IPXORG1-150 :IPYFAC=IPYORG2+250
4670 IPXA=200+2190 :IPYA=220+1200 :IPXB=200+2190 :IPYB=220+600
4680 IPXEG=IPXORG1-150 :IPYEG=IPYORG2+150
      :IPXEX=200+2120 :IPYEX=220+600
4690 IST=60 :IQT=50 :IS10=50 :IQ10=30
4700 ISEXP=30 :IQEXP=18 :ISL=50 :IQL=30
4710 CLS 1 :COLOR 6 :LOCATE 26,1 :PRINT "FORMAT SELECTION" :COLOR 5
4720 LOCATE 21,3 :PRINT "ID  ITEM" :COLOR 0
4730 LOCATE 21,5 :PRINT "0  Dot with  Axes"
4740 LOCATE 21,7 :PRINT "1  Line with  Axes"
4750 LOCATE 21,9 :PRINT "2  Dot without Axes"
4760 LOCATE 21,11:PRINT "3  Line without Axes"
4770 LOCATE 21,14:PRINT "4  No Plot"
4780 COLOR@ (21,5)-(21,14),2 :COLOR@ (26,5)-(50,14),5 :COLOR 5
4790 LOCATE 11,16 :PRINT "Type ID # to choose ITEM ;Press RETURN"
4800 *LOOP301 :A$=INKEY$
4810 IF A$="" THEN *LOOP301
4820 IOP5=VAL(A$)
4830 IOP51=IOP5 MOD 2 :IOP52=IOP5 ¥ 2
4840 IF IOP5>4 THEN *PLOTTER ELSE
      IF IOP5=4 THEN RETURN ELSE IF IOP51=1 THEN *JUMP23
4850 CONSOLE 3,22,0,1 :CLS 1 :CONSOLE 0,25,0,1 :COLOR 0
4860 LOCATE 21,4 :PRINT "0  E"
4870 LOCATE 21,6 :PRINT "1
4880 LOCATE 21,8 :PRINT "2  ~"
4890 LOCATE 21,10:PRINT "3  ""
4900 LOCATE 21,12:PRINT "4  ""
4910 COLOR@ (21,4)-(21,12),2 :COLOR@ (26,4)-(40,12),5 :COLOR 5
4920 LOCATE 11,14 :INPUT "Type ID # to choose ITEM ";IDOT
4930 IF (IDOT=3) OR (IDOT=4) THEN IDOT=7-IDOT
4940 IF (IDOT=0) OR (IDOT=1) THEN IDOT=IDOT*4+1
4950 IF IDOT=2 THEN IDOT=7
4960 IF SIG(ITOT)=0 THEN *JUMP24
4970 *JUMP23 :CONSOLE 3,22,0,1 :CLS 1 :CONSOLE 0,25,0,1 :COLOR 0
4980 LOCATE 21,4 :PRINT "0  -----"
4990 LOCATE 21,6 :PRINT "1  - - - -"
5000 LOCATE 21,8 :PRINT "2  -%-%-%"
5010 LOCATE 21,10:PRINT "3  -%-%-%"
5020 COLOR@ (21,4)-(21,10),2 :COLOR@ (26,4)-(40,10),5 :COLOR 5
5030 LOCATE 11,12 :INPUT "Type ID # to choose ITEM ";LINNE
5040 IF LINNE=0 THEN LINNE=-1
5050 LINNE=LINNE*2+2
5060 *JUMP24 :LOCATE 21,16 :INPUT "Pen # ";IPEN
5070 IPXTIT=(IPXORG1+IPXORG2)*.5-((LEN(XTIT$)-1)*IQT+IST*.57)*.5
5080 IPYTIT=(IPYORG1+IPYORG2)*.5-((LEN(YTIT$)-1)*IQT+IST*.57)*.5
5090 PXT=(IPXORG2-IPXORG1)/NXT
5100 FOR I=1 TO NXT+1
5110 IPXLAB(I)=IPXORG2-PXT*(I-1)
5120 XLAB(I)=XORG2-XT*(I-1)
5130 NEXT I
5140 PYT=(IPYORG2-IPYORG1)/NYT
5150 FOR I=1 TO NYT+1
5160 IPYLAB(I)=IPYORG2-PYT*(I-1)
5170 YLAB(I)=YORG2-YT*(I-1)
5180 NEXT I
5190 A=PXT/XT :B=PYT/YT

```

```

5200 FOR I=1 TO ITOT
5210 IF IOP11=1 THEN *JUMP25
5220 IX(I)=IPXORG2+(T(I)-XORG2)*A
5230 GOTO *JUMP26
5240 *JUMP25 :IF T(I)<=0 THEN IX(I)=IPXORG1 :GOTO *JUMP26
5250 IX(I)=IPXORG2+(LOG(T(I))/C1-XORG2)*A
5260 *JUMP26 :IF IOP12=1 THEN *JUMP27
5270 IY(I)=IPYORG2+(P(I)-YORG2)*B
5280 GOTO *JUMP28
5290 *JUMP27 :IF P(I)<=0 THEN IY(I)=IPYORG1 :GOTO *JUMP28
5300 IY(I)=IPYORG2+(LOG(P(I))/C1-YORG2)*B
5310 *JUMP28 :NEXT I
5320 PRINT@ 16;"^260,70,¥-260,-70,Z1750,2700,M",
        IPXORG2,IPYORG1,",J1,L0"
5330 IF IOP52=1 THEN *JUMP61
5340 FOR I=2 TO NXT+1
5350 PRINT@ 16;"D",IPXLAB(I),IPYORG1,CHR$(3),"E0,20,0,-20",CHR$(3)
5360 NEXT I
5370 FOR I=NYT TO 1 STEP -1
5380 PRINT@ 16;"D",IPXORG1,IPYLAB(I),CHR$(3),"E20,0,-20,0",CHR$(3)
5390 NEXT I
5400 FOR I=NXT TO 1 STEP -1
5410 PRINT@ 16;"D",IPXLAB(I),IPYORG2,CHR$(3),"E0,-20,0,20",CHR$(3)
5420 NEXT I
5430 FOR I=2 TO NYT+1
5440 PRINT@ 16;"D",IPXORG2,IPYLAB(I),CHR$(3),"E-20,0,20,0",CHR$(3)
5450 NEXT I
5460 IA10=IQ10+IS10*.57
5470 FOR I=IOPX4 TO NXT+1 STEP IOPX3
5480 IF IOP11=1 THEN *JUMP29
5490 A=LEN(STR$(XLAB(I)))
5500 PXL=IQL*A*.5
5510 PRINT@ 16;"M",IPXLAB(I)-PXL,IPYORG1+75-ISL-20,
        ",S",ISL,"Q",IQL,"P"
5520 PRINT@ 16;XLAB(I)
5530 PRINT@ 16;CHR$(3)
5540 GOTO *JUMP30
5550 *JUMP29 :A=IPYORG1-ISEXP-20
5560 PRINT@ 16;"M",IPXLAB(I)-IA10,A-IS10,"S",IS10,"Q",IQ10,"P10"
5570 PRINT@ 16;CHR$(3)
5580 PRINT@ 16;"M",IPXLAB(I),A+IS10,"S",ISEXP,"Q",IQEXP,"P"
5590 PRINT@ 16;XLAB(I)
5600 PRINT@ 16;CHR$(3)
5610 *JUMP30 :NEXT I
5620 FOR I=IOPY4 TO NYT+1 STEP IOPY3
5630 A=LEN(STR$(YLAB(I)))
5640 IF A>AA THEN AA=A
5650 NEXT I
5660 FOR I=IOPY4 TO NYT+1 STEP IOPY3
5670 IF IOP12=1 THEN *JUMP31
5680 PYL=IQL*AA
5690 PRINT@ 16;"M",IPXORG1-PYL-20,IPYLAB(I)+75-ISL*.5,
        ",S",ISL,"Q",IQL,"P"
5700 PRINT@ 16;YLAB(I)
5710 PRINT@ 16;CHR$(3)
5720 GOTO *JUMP32

```



```

5730 *JUMP31 :A=IPXORG1-IQEXP*AA-10
5740 PRINT@ 16;"M",A-IA10,IPYLAB(I)-IS10,"S",IS10,"Q",IQ10,"P10"
5750 PRINT@ 16;CHR$(3)
5760 PRINT@ 16;"M",A,IPYLAB(I)+IS10,"S",ISEXP,"Q",IQEXP,"P"
5770 PRINT@ 16;YLAB(I)
5780 PRINT@ 16;CHR$(3)
5790 *JUMP32 :NEXT I
5800 A=IPYORG1-IST-70+90
5810 IF IOP11=1 THEN A=A-ISEXP-IS10 ELSE A=A-ISL
5820 PRINT@ 16;"M",IPXTIT,A,"S",IST,"Q",IQT,"P"
5830 PRINT@ 16;XTIT$
5840 PRINT@ 16;CHR$(3)
5850 A=IPXORG1-IST-85
5860 IF IOP12=1 THEN A=A-AA*IQEXP-IA10 ELSE A=A-AA*IQL
5870 PRINT@ 16;"M",A,IPYTIT,"S",IST,"Q",IQT,"R900,P"
5880 PRINT@ 16;YTIT$
5890 PRINT@ 16;CHR$(3),"A"
5900 PRINT@ 16;"M",IPXDATE,IPYDATE,"S",IST,"Q",IQT,"P"
5910 PRINT@ 16;"Date "+DATTE$
5920 PRINT@ 16;CHR$(3),"A"
5930 PRINT@ 16;"M",IPXSAM,IPYSAM,"S",IST,"Q",IQT,"P"
5940 PRINT@ 16;"Sample "+SAMPLE$
5950 PRINT@ 16;CHR$(3),"A"
5960 PRINT@ 16;"M",IPXFAC,IPYFAC,"S",IST,"Q",IQT,"P"
5970 PRINT@ 16;"File "+FILE$
5980 PRINT@ 16;CHR$(3),"A"
5990 PRINT@ 16;"M",IPXEG,IPYEG,"S",IST,"Q",IQT,"P"
6000 PRINT@ 16;"Comment "+COMMEN$
6010 PRINT@ 16;CHR$(3),"A"
6020 *JUMP61 :PRINT@ 16;"¥",IPXORG1,IPYORG1,"Z",IPXORG2,IPYORG2
6030 IF IOP5=1 THEN *JUMP33
6040 FOR I=1 TO ITOT
6050 PRINT@ 16;"M",IX(I),IY(I),"S20,N",IDOT,"A"
6060 NEXT I
6070 GOTO *JUMP34
6080 *JUMP33
6090 PRINT@ 16;"M",IX(1),IY(1),"L",LINNE
6100 FOR I=2 TO ITOT
6110 PRINT@ 16;"D",IX(I),IY(I)
6120 NEXT I
6130 PRINT@ 16;CHR$(3),"A"
6140 *JUMP34 :PRINT@ 16;"¥-260,-70,Z2700,1750"
6150 RETURN
6160 '
6170 '----- DATA STORAGE -----
6180 *STOREDATA :CLS 1
6190 COLOR 6:LOCATE 10,0
6200 PRINT "*** STORE THE DATA ***"
6210 COLOR 6:LOCATE 40,2 :PRINT "1-----8"
6220 LOCATE 40,3 :PRINT DATE$
6230 *JUMP106
      :LOCATE 10,3 :PRINT "DATE          (8 CHARACTERS)"
      :COLOR 7:LOCATE 40,3 :LINE INPUT DATTE$
6240 IF DATTE$="" THEN DATTE$=DATE$ :LOCATE 40,3 :PRINT DATTE$
6250 IF LEN(DATTE$)>8 THEN *JUMP106
6260 COLOR 6:LOCATE 40,5 :PRINT "1-----20"

```

```

6270 LOCATE 40,6 :SAMPLE2$=SAMPLE$ :PRINT SAMPLE2$
6280 *JUMP107
      :LOCATE 10,6 :PRINT "SAMPLE NAME(20CHARACTERS)"
      :COLOR 7:LOCATE 40,6 :LINE INPUT SAMPLE$
6290 IF SAMPLE$="" THEN SAMPLE$=SAMPLE2$ :LOCATE 40,6 :PRINT SAMPLE$
6300 IF LEN(SAMPLE$)>20 THEN *JUMP107
6310 COLOR 6:LOCATE 40,8 :PRINT "1-----10"
6320 LOCATE 40,9 :FILE2$=LEFT$(SAMPLE$,10) :PRINT FILE2$
6330 *JUMP104
      :LOCATE 10,9 :PRINT "FILE NAME (10 CHARACTERS)"
      :COLOR 7:LOCATE 40,9 :LINE INPUT FILE$
6340 IF FILE$="" THEN FILE$=FILE2$ :LOCATE 40,9 :PRINT FILE$
6350 IF LEN(FILE$)>10 THEN *JUMP104
6360 *JUMP105
      :COLOR 6:LOCATE 40,11:PRINT "1-----20"
6370 LOCATE 40,12:COMMEN2$=COMMEN$ :PRINT COMMEN2$
6380 LOCATE 10,12:PRINT "COMMENT (20 CHARACTERS)"
      :COLOR 7:LOCATE 40,12:LINE INPUT COMMEN$
6390 IF COMMEN$="" THEN COMMEN$=COMMEN2$ :LOCATE 40,12:PRINT COMMEN$
6400 IF LEN(COMMEN$)>20 THEN *JUMP105
6410 GOSUB *MYSAVE : RETURN
6420 '
6430 ' ----- READ OUT "TSINDEX" -----
6440 *READFILE
6450 OPEN DV$+"TSINDEX" AS #1
6460 FIELD #1,2 AS A$
6470 GET #1,1
6480 FTOT1=CVI(A$)
6490 IF FTOT1=0 THEN CLOSE #1 :RETURN
6500 'DIM A1$(FTOT1+1)
6510 I=0
6520 *JUMP135 :I=I+1
6530 LAST=FTOT1-12*(I-1)
6540 IF LAST>12 THEN LAST1=12 ELSE LAST1=LAST
6550 FIELD #1,240 AS A$
6560 GET #1
6570 FOR J=1 TO LAST1
6580 A1$(12*(I-1)+J)=MID$(A$,20*J-19,20)
6590 NEXT J
6600 IF LAST1<>LAST THEN *JUMP135
6610 CLOSE #1
6620 RETURN
6630 ' ----- WRITE DOWN "TSINDEX" -----
6640 *WRITEFILE
6650 OPEN DV$+"TSINDEX" AS #1
6660 FIELD #1,2 AS A$
6670 LSET A$=MKI$(FTOT1)
6680 PUT #1,1
6690 I=0
6700 *JUMP136 :I=I+1
6710 LAST=FTOT1-12*(I-1)
6720 IF LAST>12 THEN LAST1=12 ELSE LAST1=LAST
6730 FIELD #1,240 AS A$
6740 B$=""
6750 FOR J=1 TO LAST1
6760 B$=B$+A1$(12*(I-1)+J)

```



```

6770 NEXT J
6780 LSET A$=B$
6790 PUT #1
6800 IF LAST1<>LAST THEN *JUMP136
6810 CLOSE #1
6820 RETURN
6830 ' ----- STORE THE DATA -----
6840 *STOREFILE
6850 IF ITOT<=29 THEN NEED=1 ELSE NEED=((ITOT-29) ¥ 63)+2
6860 ' GOSUB *READFILE
6870 I=0
6880 *JUMP138 :I=I+1
6890 IF I>FTOT1 THEN IAD1=FTOT1+1 :GOTO *JUMP139
6900 A$=LEFT$(A1$(I),10)
6910 IF FILE$>A$ THEN *JUMP138 ELSE
      IF FILE$<A$ THEN IAD1=I ELSE IRE=1
      :COLOR 2: LOCATE 21,20: PRINT FILE$;" IS ALREADY EXIST !"
      :INPUT ZZZZ : RETURN
6920 *JUMP139 :OPEN DV$+"TSCONTEN" AS #1
6930 FIELD #1,2 AS A$
6940 GET #1,1
6950 FTOT2=CVI(A$)
6960 FIELD #1,250 AS A$
6970 IF FTOT2=0 THEN IAD2=1 :IADD=1 :GOTO *JUMP140
6980 I=0
6990 *JUMP141 :I=I+1
7000 LAST=FTOT2-5*(I-1)
7010 IF LAST>5 THEN LAST1=5 ELSE LAST1=LAST
7020 GET #1
7030 FOR J=1 TO LAST1
7040 K=5*(I-1)+J
7050 A2$=MID$(A$,50*J-49,50)
7060 A3$=LEFT$(A2$,1)
7070 IF ((A3$=" ") AND (IAD2<>0)) OR
      ((A3$<>" ") AND (IAD2=0)) THEN *JUMP142
7080 IF (A3$=" ") AND (IAD2=0) THEN *JUMP143
7090 IST=VAL(MID$(A2$,42,4))
7100 IF NEED<=IST-IEN-1 THEN IADD=IEN+1 :GOTO *JUMP140
7110 IST=0 :IEN=0 :IAD2=0 :GOTO *JUMP142
7120 *JUMP143 :IEN=VAL(MID$(A4$,47,4)) :IAD2=K
7130 *JUMP142 :A4$=A2$ :NEXT J
7140 IF LAST1<>LAST THEN *JUMP141
7150 IF IAD2<>0 THEN IADD=IEN+1 ELSE IADD=VAL(MID$(A2$,47,4))+1
      :IAD2=FTOT2+1
7160 *JUMP140 :IR1=IAD2 ¥ 5
7170 IR2=IAD2 MOD 5
7180 IF IR2<>0 THEN IR1=IR1+1 ELSE IR2=5
7190 IF (IR2=1) AND (IAD2=FTOT2+1) THEN B$=SPACE$(50)
      :GOTO *JUMP137
7200 GET #1,IR1+1
7210 B$=A$
7220 *JUMP137 :MID$(B$,50*IR2-49,11)=FILE$
7230 MID$(B$,50*IR2-38,9)=DATTE$
7240 MID$(B$,50*IR2-29,20)=COMMEN$
7250 MID$(B$,50*IR2-9,5)=STR$(IADD)
7260 MID$(B$,50*IR2-4,5)=STR$(IADD+NEED-1)

```

```

7270 LSET A$=B$
7280 PUT #1,IR1+1
7290 FIELD #1,2 AS A$
7300 IF IAD2=FTOT2+1 THEN FTOT2=FTOT2+1
7310 LSET A$=MKI$(FTOT2)
7320 PUT #1,1
7330 CLOSE #1
7340 A$=SPACE$(20)
7350 MID$(A$,1,10)=FILE$
7360 MID$(A$,11,5)=STR$(IADD)
7370 MID$(A$,16,5)=STR$(IADD+NEED-1)
7380 FOR I=FTOT1+1 TO IAD1 STEP -1
7390 IF I=IAD1 THEN A1$(I)=A$ ELSE A1$(I)=A1$(I-1)
7400 NEXT I
7410 FTOT1=FTOT1+1
7420 GOSUB *WRITEFILE
7430 OPEN DV$+"TSDATA" AS #1
7440 FIELD #1,10AS F1$,8AS F2$,20AS F3$,20AS F4$,20AS F5$,
      20AS F6$,20AS F7$,16AS F8$,116AS F9$
7450 LSET F1$=FILE$
7460 LSET F2$=DATTE$
7470 LSET F3$=SAMPLE$
7480 LSET F4$=COMMEN$
7490 LSET F5$=DT$
7500 LSET F6$=IT$
7510 LSET F7$=BOS$
7520 LSET F8$=MKS$(VSTA)+MKS$(VSTO)+MKS$(VSTE)+MKS$(ITOT)
7530 IF ITOT<29 THEN LAST=ITOT ELSE LAST=29
7540 B$=""
7550 FOR I=1 TO LAST
7560 B$=B$+MKS$(P(I))
7570 NEXT I
7580 LSET F9$=B$
7590 PUT #1,IADD
7600 IF LAST=ITOT THEN *JUMP433
7610 I=0
7620 *JUMP144 :I=I+1
7630 LAST=ITOT-29-63*(I-1)
7640 IF LAST>63 THEN LAST1=63 ELSE LAST1=LAST
7650 FIELD #1,252 AS A$
7660 B$=""
7670 FOR J=1 TO LAST1
7680 B$=B$+MKS$(P(29+63*(I-1)+J))
7690 NEXT J
7700 LSET A$=B$
7710 PUT #1
7720 IF LAST1<>LAST THEN *JUMP144
7730 *JUMP433 :CLOSE #1
7740 RETURN
7750 ' ----- FIND A DATA IN "TSINDEX" -----
7760 *FINDFILE
7770 IRE=0
7780 GOSUB *READFILE
7790 I=0
7800 *JUMP145 :I=I+1
7810 IF I>FTOT1 THEN IRE=1 :RETURN

```



```

7820 A$=LEFT$(A1$(I),10)
7830 IF FILE$>A$ THEN *JUMP145 ELSE
      IF FILE$=A$ THEN IAD1=I ELSE IRE=1 :RETURN
7840 IST=VAL(MID$(A1$(IAD1),12,4))
7850 IEN=VAL(MID$(A1$(IAD1),17,4))
7860 RETURN
7870 ' ----- DELETE A DATA -----
7880 *DELFILE
7890 GOSUB *FINDFILE
7900 IF IRE=1 THEN RETURN
7910 FOR I=IAD1 TO FTOT1-1
7920 A1$(I)=A1$(I+1)
7930 NEXT I
7940 FTOT1=FTOT1-1
7950 GOSUB *WRITEFILE
7960 OPEN DV$+"TSCONTEN" AS #1
7970 FIELD #1,2 AS A$
7980 GET #1,1
7990 FTOT2=CVI(A$)
8000 I=0
8010 *JUMP146 :I=I+1
8020 LAST=FTOT2-5*(I-1)
8030 IF LAST>5 THEN LAST1=5 ELSE LAST1=LAST
8040 FIELD #1,250 AS A$
8050 GET #1
8060 FOR J=1 TO LAST1
8070 K=5*(I-1)+J
8080 A2$=MID$(A$,50*J-49,10)
8090 IF A2$=FILE$ THEN *JUMP147
8100 NEXT J
8110 IF LAST1<>LAST THEN *JUMP146
8120 IRE=2 :CLOSE #1
8130 RETURN
8140 *JUMP147 :B$=A$
8150 MID$(B$,50*J-49,50)=SPACE$(50)
8160 LSET A$=B$
8170 PUT #1,I+1
8180 IF (LAST1<>LAST) OR (J<>LAST1) THEN *JUMP148
8190 FIELD #1,2 AS A$
8200 FTOT2=FTOT2-1
8210 LSET A$=MKI$(FTOT2)
8220 *JUMP148 :CLOSE #1
8230 OPEN DV$+"TSDATA" AS #1
8240 FIELD #1,255 AS F1$,1 AS F2$
8250 LSET F1$=SPACE$(255)
8260 LSET F2$=SPACE$(1)
8270 FOR I=IST TO IEN
8280 PUT #1,I
8290 NEXT I
8300 CLOSE #1
8310 RETURN
8320 ' ----- LOADING THE DATA -----
8330 *LOADFILE
8340 A=LEN(FILE$)
8350 FILE$=FILE$+SPACE$(10-A)
8360 GOSUB *FINDFILE

```

```

8370 'INPUT IST
8380 IF IRE=1 THEN COLOR 2:LOCATE 21,24
      :PRINT LEFT$(FILE$,A);" IS NOT EXIST !"
8390 OPEN DV$+"TSDATA" AS #1
8400 FIELD #1,10AS F1$,8AS F2$,20AS F3$,20AS F4$,20AS F5$,
      20AS F6$,20AS F7$,16AS F8$,116AS F9$

8410 GET #1,IST
8420 FILE$=F1$
8430 FILE$=LEFT$(FILE$,10)
8440 DATTE$=F2$
8450 DATTE$=LEFT$(DATTE$,8)
8460 SAMPLE$=F3$
8470 SAMPLE$=LEFT$(SAMPLE$,20)
8480 COMMEN$=F4$
8490 COMMEN$=LEFT$(COMMEN$,20)
8500 DT$=F5$
8510 A=INSTR(DT$," ")-1
8520 IF A>0 THEN DT$=LEFT$(DT$,A)
8530 IT$=F6$
8540 A=INSTR(IT$," ")-1
8550 IF A>0 THEN IT$=LEFT$(IT$,A)
8560 BOS$=F7$
8570 A=INSTR(BOS$," ")-1
8580 IF A>0 THEN BOS$=LEFT$(BOS$,A)
8590 VSTA=CVS(MID$(F8$,1,4))
8600 VSTO=CVS(MID$(F8$,5,4))
8610 VSTE=CVS(MID$(F8$,9,4))
8620 ITOT=CVS(MID$(F8$,13,4))
8630 BOS=VAL(BOS$)
8640 '
8650 IF BOS=0 THEN 8760
8660 '
8670 FOR I=1 TO CINT(ITOT/2)
8680 T(I)=VSTA+(I-1)*VSTE
8690 NEXT I
8700 K=0
8710 FOR I=CINT(ITOT/2)+1 TO ITOT
8720 K=K+1 :T(I)=VSTO-(K-1)*VSTE
8730 NEXT I
8740 GOTO 8920
8750 '
8760 FOR I=1 TO CINT(ITOT/4)
8770 T(I)=VSTA+(I-1)*VSTE
8780 NEXT I
8790 K=0
8800 FOR I=CINT(ITOT/4)+1 TO CINT(ITOT/2)
8810 K=K+1 :T(I)=VSTO-(K-1)*VSTE
8820 NEXT I
8830 K=0
8840 FOR I=CINT(ITOT/2)+1 TO CINT(ITOT/2+ITOT/4)
8850 K=K+1 :T(I)=-VSTA-(K-1)*VSTE
8860 NEXT I
8870 K=0
8880 FOR I=CINT(ITOT/2+ITOT/4)+1 TO ITOT
8890 K=K+1 :T(I)=-VSTO+(K-1)*VSTE
8900 NEXT I

```



```

8910 '
8920 IF ITOT<29 THEN LAST=ITOT ELSE LAST=29
8930 FOR I=1 TO LAST
8940 J=4*I-3
8950 P(I)=CVS(MID$(F9$,J,4))
8960 NEXT I
8970 IF LAST=ITOT THEN *JUMP149
8980 I=0
8990 *JUMP150 :I=I+1
9000 LAST=ITOT-29-63*(I-1)
9010 IF LAST>63 THEN LAST1=63 ELSE LAST1=LAST
9020 FIELD #1,252 AS A1$
9030 GET #1
9040 FOR J=1 TO LAST1
9050 K=4*J-3
9060 P(29+63*(I-1)+J)=CVS(MID$(A1$,K,4))
9070 NEXT J
9080 IF LAST1<>LAST THEN *JUMP150
9090 *JUMP149 :CLOSE #1
9100 RETURN
9110 ' ----- 2.3 FILE MANAGEMENT BRANCH -----
9120 *FILEMANA :CONSOLE 0,24,0,1 :CLS 3 :COLOR 6
9130 LOCATE 26,0 :PRINT "FILE MANAGEMENT TASKS" :COLOR 5
9140 LOCATE 21,2 :PRINT "0 LIST INDEX OF DATA FILES ON CRT"
9150 LOCATE 21,3 :PRINT "1 PRINT INDEX OF DATA FILES"
9160 LOCATE 21,4 :PRINT "2 LIST CONTENTS OF DATA FILES ON CRT"
9170 LOCATE 21,5 :PRINT "3 PRINT CONTENTS OF DATA FILES"
9180 LOCATE 21,6 :PRINT "4 DELETE A DATA FILE"
9190 LOCATE 21,7 :PRINT "5 GO TO TASK SELEKTION"
9200 LOCATE 26,9 :PRINT "PRESS YOUR CHOICE NUMBER"
9210 COLOR 7:LOCATE 29,10:PRINT "AND PRESS ";:COLOR 2:PRINT "G ";
:COLOR 7:PRINT "TO ESCAPE IN ";:COLOR 2:PRINT "MODE 1-4"
9220 COLOR@ (21,2)-(21,7),2 :COLOR@ (26,9)-(60,9),2 :COLOR 5
9230 *LOOP200 :C$=INKEY$
9240 IF C$="" THEN *LOOP200
9250 IF C$="0" THEN GOSUB *LIST1 :GOTO *FILEMANA
9260 IF C$="1" THEN GOSUB *LIST1L :GOTO *FILEMANA
9270 IF C$="2" THEN GOSUB *LIST2 :GOTO *FILEMANA
9280 IF C$="3" THEN GOSUB *LIST2L :GOTO *FILEMANA
9290 IF C$="5" OR C$=CHR$(13) THEN RETURN
9300 IF C$<>"4" THEN *FILEMANA
9310 CLS 3 :LOCATE 21,9
:PRINT "INSERT DATA DISKETTE INTO RIGHT SLOT"
9320 COLOR 7:LOCATE 21,11
:PRINT "FILE NAME";:COLOR 5:PRINT " TO BE DELETED ";
:COLOR 7:LINE INPUT FILE$
9330 A=LEN(FILE$)
9340 FILE$=FILE$+SPACE$(10-A)
9350 GOSUB *DELFILE
9360 COLOR 2 :LOCATE 21,24 :PRINT LEFT$(FILE$,A);
9370 IF IRE<>1 THEN PRINT " WAS NOW DELETED" :GOTO *FILEMANA
9380 PRINT " IS NOT EXIST !" :GOTO *FILEMANA
9390 ' ----- LIST INDEX OF DATA ON CRT -----
9400 *LIST1
9410 IOP6=1
9420 GOSUB *LISTCOM1

```

```

9430 RETURN
9440 ' ----- PRINT INDEX OF DATA BY PRINTER -----
9450 *LIST1L
9460 IOP6=2
9470 GOSUB *LISTCOM1
9480 RETURN
9490 ' ----- COMMON ROUTINE OF LIST OR PRINT INDEX OF DATA -----
9500 *LISTCOM1
9510 GOSUB *READFILE
9520 IF IOP6=1 THEN PRINT "FTOT1 = ";FTOT1 ELSE
      IF IOP6=2 THEN LPRINT "FTOT1 = ";FTOT1
9530 IR1=FTOT1 / 3
9540 IR2=FTOT1 MOD 3
9550 FOR I=1 TO IR1
9560 IF IOP6=1 THEN PRINT A1$(3*I-2);SPACE$(5);A1$(3*I-1);
      SPACE$(5);A1$(3*I)
9570 IF IOP6=2 THEN LPRINT A1$(3*I-2);SPACE$(5);A1$(3*I-1);
      SPACE$(5);A1$(3*I)
9580 NEXT I
9590 FOR I=1 TO IR2
9600 IF IOP6=1 THEN PRINT A1$(3*IR1+I);SPACE$(5) ELSE
      IF IOP6=2 THEN LPRINT A1$(3*IR1+I);SPACE$(5)
9610 NEXT I
9620 *HALT10 :D$=INKEY$ :IF D$<>"G" THEN *HALT10 ELSE RETURN
9630 ' ----- LIST CONTENTS OF DATA FILE ON CRT -----
9640 *LIST2
9650 IOP7=1
9660 GOSUB *LISTCOM2
9670 RETURN
9680 ' ----- PRINT CONTENTS OF DATA FILE BY PRINTER -----
9690 *LIST2L
9700 IOP7=2
9710 GOSUB *LISTCOM2
9720 RETURN
9730 ' -- COMMON ROUTINE OF LIST OR PRINT CONTENTS OF DATA FILE --
9740 *LISTCOM2
9750 OPEN DV$+"TSCONTEN" AS #1
9760 FIELD #1,2 AS A$
9770 GET #1,1
9780 FTOT2=CVI(A$)
9790 IF IOP7=1 THEN PRINT "FTOT2 = ";FTOT2 ELSE
      IF IOP7=2 THEN LPRINT "FTOT2 = ";FTOT2
9800 IR1=FTOT2 / 5
9810 IR2=FTOT2 MOD 5
9820 FIELD #1,250 AS A$
9830 FOR I=1 TO IR1
9840 GET #1
9850 FOR J=1 TO 5
9860 IF IOP7=1 THEN PRINT MID$(A$,50*J-49,50) ELSE
      IF IOP7=2 THEN LPRINT MID$(A$,50*J-49,50)
9870 NEXT J
9880 NEXT I
9890 GET #1
9900 FOR I=1 TO IR2
9910 IF IOP7=1 THEN PRINT MID$(A$,50*I-49,50) ELSE
      IF IOP7=2 THEN LPRINT MID$(A$,50*I-49,50)

```



```

9920 NEXT I
9930 CLOSE #1
9940 *HALT20:D$=INKEY$ :IF D$<>"G" THEN *HALT20 ELSE RETURN
9950 '----- LOADED DATA DISPLY -----
9960 *LDDISP: CONSOLE 0,25,0,1 :CLS 1: COLOR 6
9970 LOCATE 20,0 :PRINT "***** LOADED DATA *****"
9980 COLOR 5:LOCATE 20,2 :PRINT "FILE NAME":COLOR 7:LOCATE 41,2
:PRINT FILE$
9990 COLOR 5:LOCATE 20,3 :PRINT "COMMENT" :COLOR 7:LOCATE 41,3
:PRINT COMMEN$
10000 COLOR 5:LOCATE 20,5 :PRINT "DATE":COLOR 7:LOCATE 41,5
:PRINT DATTE$
10010 COLOR 5:LOCATE 20,6 :PRINT "SAMPLE NAME":COLOR 7:LOCATE 41,6
:PRINT SAMPLE$
10020 COLOR 5:LOCATE 20,8 :PRINT "DELAY TIME":COLOR 7:LOCATE 41,8
:PRINT DT$
10030 COLOR 5:LOCATE 20,9 :PRINT "INTEGRATION" :COLOR 7:LOCATE 41,9
:PRINT IT$
10040 COLOR 5:LOCATE 20,10:PRINT "POLARITY"
:COLOR 7:LOCATE 41,10:IF BOS=0 THEN PRINT "BOTH SIDE" ELSE
IF BOS=1 THEN PRINT "SINGLE SIDE"
10050 *JUMP1000:COLOR 7
:LOCATE 20,12 :PRINT "1      HARDCOPY & CONTINUE"
10060 LOCATE 20,13 :PRINT "2 NOT HARDCOPY & CONTINUE"
10070 LOCATE 21,15 :COLOR 2 :PRINT "PRESS YOUR CHOICE NUMBER"
:COLOR 5
10080 COLOR@ (20,12)-(20,13),2
10090 *LOOP1000:C$=INKEY$
10100 IF C$="" THEN *LOOP1000
10110 IF C$="2" OR C$=CHR$(13)THEN *ESCAPE1000
10120 IF C$<>"1" THEN *JUMP1000
10130 CONSOLE 17,20,0,1 :CLS 1 :CONSOLE 0,25,0,1 :COPY
10140 *ESCAPE1000
10150 RETURN
10160 '----- TSCONTEN INPUT -----
10170 *CONTEN
10180 OPEN DV$+"TSCONTEN" AS #1
10190 FIELD #1,2 AS A$
10200 GET #1,1
10210 FTOT2=CVI(A$)
10220 IR1=FTOT2 ¥ 5:IR2=FTOT2 MOD 5
10230 FIELD #1,250 AS A$
10240 FOR I=1 TO IR1
10250 GET #1
10260 FOR J=1 TO 5
10270 II=5*I+J-6:BB$(II)=MID$(A$,50*J-49,50)
10280 NEXT J :NEXT I
10290 IF IR2=<0 THEN GOTO 10340
10300 GET #1
10310 FOR I=1 TO IR2
10320 II=5*IR1+I-1:BB$(II)=MID$(A$,50*I-49,50)
10330 NEXT I
10340 CLOSE #1
10350 RETURN
10360 '----- FILE INPUT -----
10370 *FILERE$

```

```

10380 CLS 1
10390 GOSUB *CONTEN
10400 WANT=FTOT2-1:IMIN=FTOT2-16
10410 COLOR 4:LOCATE 20,0: PRINT " TS DATA FILES "
10420 COLOR 7:LOCATE 50,0: PRINT "FTOT=";FTOT2
10430 COLOR 4:LOCATE 10,1 : PRINT STRING$(58,"-")
10440 COLOR 4:LOCATE 10,18: PRINT STRING$(58,"-")
10450 *LP1 :COLOR 3 : LOCATE 15,19 : PRINT"FILES LIST =";
      :FILES DV$
10460 LOCATE 0,21:FOR I=1 TO 3:PRINT SPC(79):NEXT I
10470 COLOR 2:LOCATE 20,22:PRINT " PLEASE SELECT"
10480 *LP2 :COLOR 5
10490 FOR I=IMIN TO IMIN+15
10500 IF I=>0 THEN LOCATE 15,2+I-IMIN: PRINT BB$(I)
10510 NEXT I
10520 IF WANT<0 THEN WANT=FTOT2-1
10530 COLOR 6:LOCATE 15,2+WANT-IMIN:PRINT BB$(WANT)
10540 WHILE W$<>"":W$=INKEY$ :WEND
10550 *LP3:W$=INKEY$
10560 IF W$="" THEN *LP3
10570 IF W$=CHR$(30) THEN
      IF WANT=IMIN THEN WANT=WANT+15:GOTO *LP2
10580 IF W$=CHR$(30) THEN
      IF WANT>IMIN THEN WANT=WANT-1:GOTO *LP2
10590 IF W$=CHR$(31) THEN
      IF WANT=IMIN+15 THEN WANT=IMIN:GOTO *LP2
10600 IF W$=CHR$(31) THEN
      IF WANT<IMIN+15 THEN WANT=WANT+1:GOTO *LP2
10610 IF W$=CHR$(28) THEN
      IF IMIN>0 THEN IMIN=IMIN-1:WANT=WANT-1
10620 IF W$=CHR$(29) THEN
      IF IMIN+15<FTOT2-1 THEN IMIN=IMIN+1:WANT=WANT+1
10630 IF W$=CHR$(13) THEN *LP4
10640 GOTO *LP2
10650 *LP4
10660 FILE$=LEFT$(BB$(WANT),10)
10670 RETURN
10680 '
10690 '
10700 '
10710 '
10720 '
10730 '
10740 '
10750 '--- AUTO MEASUREMENT -----
10760 *MEASURE
10770 GOSUB *PARASET
10780 GOSUB *DEVICESET
10790 '
10800 I=0 :K=0
10810 *REPEAT
10820 I=I+1
10830 GOSUB *RDATA
10840 GOSUB *WDATA
10850 IF STS$ <> "L" THEN *REPEAT
10860 IF BOS=1 THEN 10950

```



```

10870 '
10880 GOSUB *DSET2
10890 *REPEAT2
10900 I=I+1
10910 GOSUB *RDATA
10920 GOSUB *WDATA
10930 IF STS$ <> "L" THEN *REPEAT2
10940 '
10950 GOSUB *DRESET
10960 '
10970 ITOT=I
10980 VSTA=PS :VSTO=PT :VSTE=PE
10990 FOR I=1 TO 10 :BEEP :NEXT I
11000 IF (BOS=0) AND (ITOT=4*(CINT((VSTO-VSTA)/VSTE)+1)) THEN 11020
11010 IF (BOS=1) AND (ITOT=2*(CINT((VSTO-VSTA)/VSTE)+1)) THEN 11020
11020 '
11030 GOTO *DISPLY
11040 '
11050 RETURN
11060 '
11070 '--- DEVICE SET -----
11080 *DEVICESET
11090 ISET IFC
11100 ISET REN
11110 CMD DELIM=0
11120 '--- 4140B SET (I) -----
11130 F=2 'function (I-V)
11140 RA=1 'I-range (AUTO)
11150 H=12 'limit auto range (1E-12A)
11160 I=3 'integration time (LONG)
11170 IT$="LONG"
11180 J=1 'filter (ON)
11190 T=1 'I-trigger (INTERNAL)
11200 A=3 'Va-mode (double staircase)
11210 B=2 'Vb-mode (OFF)
11220 L=3 'Va I-limit (1E-2A)
11230 W=1 'sweep control (auto start)
11240 '-----
11250 F$="F"+STR$(F)
11260 RA$="RA"+STR$(RA)
11270 H$="H"+STR$(H)
11280 I$="I"+STR$(I)
11290 J$="J"+STR$(J)
11300 T$="T"+STR$(T)
11310 A$="A"+STR$(A)
11320 B$="B"+STR$(B)
11330 L$="L"+STR$(L)
11340 W$="W"+STR$(W)
11350 '-----
11360 DT$=STR$(PD)+"s"
11370 BOS$=STR$(BOS)
11380 '
11390 PC1$=F$+RA$+H$+I$+J$+T$+A$+B$+L$
11400 PRINT@ 16;PC1$
11410 PRINT@ 16;PC2$
11420 PRINT@ 16;W$

```

```

11430 RETURN
11440 '
11450 '--- DEVICE SET 2 -----
11460 *DSET2
11470 FOR AAA=1 TO 5 :TTIME$=TIME$
11480 WHILE TTIME$=TIME$ :WEND :NEXT AAA
11490 PRINT@ 16;PC3$
11500 PRINT@ 16;W$
11510 RETURN
11520 '
11530 '--- DEVICE RESET -----
11540 *DRESET
11550 PRINT@ 16;"W7"
11560 IRESET REN
11570 RETURN
11580 '
11590 '--- DATA INPUT FROM 4140B -----
11600 *RDATA
11610 INPUT@ 16; A$,B$
11620 P(I)=VAL(MID$(A$,3,10))/AREA
11630 T(I)=VAL(MID$(B$,2,7))
11640 STS$=MID$(A$,1,1)
11650 RETURN
11660 '
11670 '--- PARAMETER SET -----
11680 *PARASET
11690 CONSOLE 0,25,0,1 :CLS 3
11700 COLOR 6 :LOCATE 19,0: PRINT "***** PARAMETER SET *****":
11710 COLOR 5 :LOCATE 23,1: PRINT "FROM (V) ";PS
11720 COLOR 5 :LOCATE 23,2: PRINT "TO (V) ";PT
11730 COLOR 5 :LOCATE 23,3: PRINT "STEP (V) ";PE
11740 COLOR 5 :LOCATE 13,4: PRINT "BOTH(0) or SINGLE(1) ";BOS
11750 COLOR 5 :LOCATE 19,5: PRINT "DELAY TIME (s) ";PD
11760 COLOR 5: LOCATE 17,6: PRINT "CELL AREA (cm^2) ";AREA
11770 COLOR@ (13,4)-(33,4),4
11780 COLOR@ (19,5)-(33,5),3
11790 COLOR@ (17,6)-(33,6),2
11800 COLOR 7
11810 COLOR@ (23,1)-(33,1),7 :LOCATE 41,1 :INPUT ;C$
11820 COLOR@ (23,1)-(33,1),5
11830 IF C$<>" " THEN PS=VAL(C$) ELSE LOCATE 42,1 :PRINT PS
11840 COLOR@ (23,2)-(33,2),7 :LOCATE 41,2 :INPUT ;C$
11850 COLOR@ (23,2)-(33,2),5
11860 IF C$<>" " THEN PT=VAL(C$) ELSE LOCATE 42,2 :PRINT PT
11870 COLOR@ (23,3)-(33,3),7 :LOCATE 41,3 :INPUT ;C$
11880 COLOR@ (23,3)-(33,3),5
11890 IF C$<>" " THEN PE=VAL(C$) ELSE LOCATE 42,3 :PRINT PE
11900 *JMPPS1
11910 COLOR@ (13,4)-(33,4),7 :LOCATE 41,4 :INPUT ;C$
11920 COLOR@ (13,4)-(33,4),4
11930 IF C$<>" " THEN BOS=VAL(C$) ELSE LOCATE 42,4 :PRINT BOS
11940 IF (BOS=0) OR (BOS=1) THEN *JMPPS2 ELSE *JMPPS1
11950 *JMPPS2
11960 COLOR@ (19,5)-(33,5),7 :LOCATE 41,5 :INPUT ;C$
11970 COLOR@ (19,5)-(33,5),3
11980 IF C$<>" " THEN PD=VAL(C$) ELSE LOCATE 42,5 :PRINT PD

```



```

11990 COLOR@ (17,6)-(33,6),7 :LOCATE 41,6 :INPUT ;C$
12000 COLOR@ (17,6)-(33,6),2
12010 IF C$<>" " THEN AREA=VAL(C$) ELSE LOCATE 42,6 :PRINT AREA
12020 PH=PD
12030 IF PT-PS < 0 THEN PE=-ABS(PE)
12040 '
12050 '
12060 PS$="PS"+STR$(PS)
12070 PS2$="PS"+STR$(-PS)
12080 PT$="PT"+STR$(PT)
12090 PT2$="PT"+STR$(-PT)
12100 PE$="PE"+STR$(PE)
12110 PE2$="PE"+STR$(-PE)
12120 PH$="PH"+STR$(PH)
12130 PD$="PD"+STR$(PD)
12140 PC2$=PS$+", "+PT$+", "+PE$+", "+PH$+", "+PD$
12150 PC3$=PS2$+", "+PT2$+", "+PE2$+", "+PH$+", "+PD$
12160 '
12170 COLOR 5
12180 LOCATE 20,8:PRINT "V (V)"
12190 LOCATE 38,8:PRINT "J (A/cm^2)"
12200 LOCATE 56,1 :PRINT "STATUS"
12210 RETURN
12220 '
12230 '----- DATA DISPLAY -----
12240 *WDATA
12250 COLOR 7
12260 IF I MOD 15 = 1 AND I<>1 THEN K=K+15 :CONSOLE 9,15,0,1
      :CLS 1 :CONSOLE 0,25,0,1
12270 LOCATE 19,8+(I-K):PRINT USING "+##.## ";T(I)
12280 LOCATE 37,8+(I-K):PRINT USING "+#.##^"; P(I)
12290 LOCATE 65,1 :PRINT STS$
12300 RETURN
12310 '
12320 '
12330 '
12340 '
12350 '
12360 '---- CHANGE DIRECTLY (CHDIR) ----
12370 *DIRCH
12380 ON ERROR GOTO *ERRORTRP
12390 CLS 1 :COLOR 2 :LOCATE 7,0 :PRINT "*** CHANGE DIR ***";
12400 COLOR 3 :LOCATE 28,0 :PRINT "6Z]D C^(Z8DX= "; :FILES DV$
12410 COLOR 6 :LOCATE 0,1
12420 FOR I=1 TO 5 :PRINT SPC(79) :NEXT I :LOCATE 0,1
12430 FILES DV$+"¥"
12440 PRINT :COLOR 4
12450 PRINT "          Input a DIR name to change ";
      :COLOR 6 :INPUT DIRE1$
12460 IF DIRE1$="" THEN *ESCCH
12470 CHDIR DV$+"¥"+DIRE1$
12480 *ESCCH
12490 ON ERROR GOTO 0
12500 RETURN *DISPLY
12510 '
12520 '

```

```

12530 '---- MAKE DIRECTORY (MKDIR) ----
12540 *DIRMK
12550 CLS 1 :COLOR 4 :LOCATE 7,0 :PRINT "*** MAKE DIR ***"
12560 COLOR 3 :LOCATE 28,0 :PRINT "6Z]D C^(Z8DX= "; :FILES DV$
12570 COLOR 6 :LOCATE 0,1
12580 FOR I=1 TO 5 :PRINT SPC(79) :NEXT I :LOCATE 0,1
12590 FILES DV$+"%"
12600 PRINT :COLOR 2
12610 PRINT "          Input a new DIR name to make ";
      :COLOR 6 :INPUT DIRE1$
12620 IF DIRE1$="" THEN GOTO *ESCMK
12630 MKDIR DV$+"%" +DIRE1$ :CHDIR DV$+"%" +DIRE1$
12640 GOSUB *INITIAL
12650 *ESCMK
12660 RETURN *DISPLY
12670 '
12680 '
12690 '---- INITIALIZATION OF DATA DISKET -----
12700 *INITIAL
12710 FOR I=1 TO 3
12720 IF I=1 THEN A$="TSINDEX"
12730 IF I=2 THEN A$="TSCONTEN"
12740 IF I=3 THEN A$="TSDATA"
12750 OPEN DV$+A$ AS #1
12760 FIELD #1,2 AS A$
12770 LSET A$=MKI$(0)
12780 PUT #1,1
12790 CLOSE #1
12800 NEXT I
12810 RETURN
12820 '
12830 '----- Error trap routine -----
12840 *ERRORTRP
12850 IF ERR=76 THEN RESUME *DIRCH
12860 ON ERROR GOTO 0
12870 '
12880 '----- PRINT OUT DATA OF J-V -----
12890 *PRDATA
12900 LPRINT :LPRINT
12910           LPRINT "          V (V)          J(A/cm^2)  "
12920 FOR I=1 TO ITOT
12930 LPRINT USING "          +##.##          +#.###^ ^ ^ ^";T(I),P(I)
12940 NEXT I
12950 LPRINT :LPRINT
12960 RETURN
12970 '
12980 '----- SEARCH OF DATA -----
12990 *SDATA
13000 PRINT "Input [E] if you want to finish":PRINT
13010 INPUT "Please input a DATA of Voltage";VOLDATA$
13020 IF VOLDATA$="E" THEN RETURN
13030 FOR I=1 TO ITOT
13040 IF VAL(VOLDATA$)=T(I) THEN PRINT USING "+#.##^ ^ ^ ^";P(I)
13050 NEXT I
13060 GOTO 13010
13070 '

```



```

13080 '----- PLOTTING INPUTTED DATA-----
13090 ' (J - T^-1/4 CHARACTERISTICS)
13100 '
13110 *PLOTDATA
13120 INPUT "Input number of DATA(ITOT)";ITOT
13130 FOR I=1 TO ITOT
13140 READ T(I),P(I)
13150 T(I)=(T(I)+273)^(-1/4)
13160 NEXT I
13170 '
13180 '
13190 GOTO *DISPLY
13200 '
13210 '----Tunneling Current Plotting(J/V^2--V^(-1))
13220 *TUNNEL : JSAT=P(16)
13230 FOR I=1 TO ITOT
13240 IF T(I)=0 THEN P(I)=0 :GOTO 13280
13250 P(I)=(P(I)-JSAT)/T(I)^2
13260 IF P(I)<0 THEN P(I)=0
13270 T(I)=1/T(I)
13280 NEXT I
13290 RETURN
13300 '
13310 '----- No mean -----
13320 *K1 :RETURN
13330 *K2 :RETURN
13340 *K3 :RETURN
13350 *K4 :RETURN
13360 *K5 :RETURN
13370 *K6 :RETURN
13380 *K7 :RETURN
13390 *K9 :RETURN
13400 '
13410 '
13420 DATA 100,2.15E-4 'No.1
13430 DATA 75,1.93E-4 ' 2
13440 DATA 50,1.7E-4 '3
13450 DATA 22,1.41E-4 '4
13460 DATA 0,1.18E-4 '5
13470 DATA -25,1E-4 '6
13480 DATA -50,8.24E-5 '7
13490 'DATA 8
13500 'DATA
13510 'DATA
13520 'DATA
13530 'DATA
13540 'DATA
15000 '-----
15010 '
15020 *MYSAVE ' STORE DATA '
15030 OPEN DV$ + FILE$ AS #1
15040 WRITE #1, DATTE$;SAMPLE$;FILE$
15050 WRITE #1, COMMEN$
15060 WRITE #1, ITOT
15070 FOR I = 1 TO ITOT
15080 WRITE #1, T(1);P(1)

```

```
15090 NEXT I
15100 CLOSE #1
15110 RETURN
15120 '
15510 '
15520 *INFOTRAN
15530 PRINT #2, DATTE$
15540 PRINT #2, FILE$
15550 PRINT #2, SAMPLE$
15560 PRINT #2, COMMEN$
15570 PRINT #2, "No. of item : ";ITOT
15580 PRINT #2, "Voltage          Current"
15590 RETURN
16010 '
16020 *DATATRAN
16030 FOR I = 1 TO ITOT
16040 PRINT #2, T(I),P(I)
16050 NEXT I
16060 RETURN
16070 '
17010 '
17020 *MYLOAD
17030 OPEN DV$ + FILE$ FOR INPUT AS #1
17040 INPUT #1, DATTE$
17042 INPUT #1, SAMPLE$
17044 INPUT #1, FILE$
17050 INPUT #1, COMMEN$
17060 INPUT #1, ITOT
17070 FOR I = 1 TO ITOT
17080 INPUT #1, T(I),P(I)
17090 NEXT I
17100 CLOSE #1
17110 FOR I = 1 TO ITOT
17120 T(I) = ABS(T(I)) : P(I) = ABS(P(I))
17130 NEXT I
17140 GOSUB *MINMAX : GOSUB *XYDTMN : GOSUB *COMGRA : GOSUB *CRT
17150 GOSUB *HC      : GOTO *DISPLY
17160 RETURN
```


ภาคผนวก ค.

โปรแกรมการวัดลักษณะสมบัติของเซลล์แสงอาทิตย์ ในสภาวะได้รับพลังงานจากแสง

```

1000      '*****
1010      '*
1020      '*          SOLAR CELL PERFORMANCE          *
1030      '*
1040      '*          by 4140B pA METER HP and SOLAR SIMULATOR          *
1050      '*
1060      '*          01/08/91 edited by SUWAT SOPITPAN          *
1070      '*
1080      '*****
1090      '
1100      '
1110      '
1120      '----- MAIN PROGRAM -----
1130      GOSUB *DATASET
1140      *MAIN
1150      CONSOLE 0,25,0,1 :CLS 3
1160      COLOR 6
1170      :LOCATE 16,3 :PRINT "***** SOLAR CELL PERFORMANCE *****"
1180      COLOR 5
1190      :LOCATE 26,5 :PRINT "TASK SELECTION"
1200      LOCATE 22,7 :PRINT "1 AUTO MEASUREMENT"
1210      LOCATE 22,8 :PRINT "2 DATA TREATMENT"
1220      LOCATE 22,9 :PRINT "3 QUIT"
1230      COLOR 2:LOCATE 22,11:PRINT "PRESS YOUR CHOICE NUMBER"
1240      COLOR@ (22,7)-(22,9),2
1250      *LOOP1:C$=INKEY$
1260      IF C$="" THEN *LOOP1
1270      IF C$="1" THEN GOSUB *MEASURE
1280      IF C$="2" THEN GOSUB *DATATRE
1290      IF C$="3" THEN GOTO *BYE
1300      GOTO *MAIN
1310      *BYE : END
1320      '
1330      '----- DATA SET -----
1340      *DATASET :C1=LOG(10)
1350      DEFINT I,K-N
1360      DIM V(500),J(500)
1370      AREA=.26 :PIN=100
1380      PS=0:PT=.8:PD=.1:PE=.01
1390      DATTE$=DATE$ :USE$="N"
1400      DIM IPXLAB(50),IPYLAB(50),XLAB(50),YLAB(50)
1410      DIM P(500),T(500),IX(500),IY(500)
1420      DIM XLAB$(50),YLAB$(50),A1$(500),BB$(500),B1$(2)
1430      YTIT$="J (mA/cm^2)":XTIT$="VOLTAGE (V)"
1440      DV$="A:"
1450      RETURN

```

```

1410      '
10000  '----- AUTO MEASUREMENT ROUTINE -----
10010 *MEASURE
10020  GOSUB *INITSET
10030  GOSUB *DEVICESET
10040  I=0
10050  '
10120  WHILE J(I)>-.000001
10125  '
10130    I=I+1
10140  '
10170  GOSUB *JREAD
10180  GOSUB *WDATA
10185  '
10190  IF STS$ = "L" THEN GOTO *ITOT
10200  WEND
10210  '
10220  *ITOT : ITOT=I
10225  '
10230  GOSUB *DVCRESET
10240  GOSUB *PARACALC
10250  FOR I=1 TO ITOT :T(I)=V(I) :P(I)=J(I) :NEXT I
10260  GOTO *DISPLY
10270  '
10300  RETURN
10310  '
10330  '----- INITIAL DATA SET -----
10340  *INITSET
10350  CONSOLE 0,25,0,1 :CLS 3
10360  COLOR 6
      :LOCATE 15,0: PRINT "***** INITIAL SET *****"
10370  LOCATE 15,1: PRINT "SAMPLE NAME";SPACE$(23);SAMPLE$
10380  LOCATE 15,2: PRINT "DATE";SPACE$(30);DATTE$
10390  LOCATE 15,3: PRINT "CELL AREA (cm^2)";SPACE$(17);AREA
10400  LOCATE 15,4: PRINT "P INPUT (mW/cm^2)";SPACE$(16);PIN
10401  LOCATE 15,5: PRINT "FROM (V)";SPACE$(26);PS
10402  LOCATE 15,6: PRINT "TO (V)";SPACE$(26);PT
10403  LOCATE 15,7: PRINT "STEP (V)";SPACE$(25);PE
10404  LOCATE 15,8: PRINT "DELAY TIME (S)"SPACE$(19);PD
10405  LOCATE 15,9: PRINT "USE DIVIDER 10:1(Y/N)";SPACE$(9);USE$
10410  COLOR@ (15,1)-(31,8),5 :COLOR 7
10420  COLOR@ (15,1)-(31,1),7:LOCATE 47,1 :INPUT ;C$
10430  COLOR@ (15,1)-(31,1),5:IF C$<>" " THEN SAMPLE$=C$
10440  COLOR@ (15,2)-(31,2),7:LOCATE 47,2 :INPUT ;C$
10450  COLOR@ (15,2)-(31,2),5:IF C$<>" " THEN DATTE$=C$
10460  COLOR@ (15,3)-(31,3),7:LOCATE 47,3 :INPUT ;C$
10470  COLOR@ (15,3)-(31,3),5:IF C$<>" " THEN AREA=VAL(C$)
10480  COLOR@ (15,4)-(31,4),7:LOCATE 47,4 :INPUT ;C$
10490  COLOR@ (15,4)-(31,4),5:IF C$<>" " THEN PIN=VAL(C$)
10491  COLOR@ (15,5)-(31,5),7:LOCATE 47,5 :INPUT ;C$
10492  COLOR@ (15,5)-(31,5),5:IF C$<>" " THEN PS=VAL(C$)
10493  COLOR@ (15,6)-(31,6),7:LOCATE 47,6 :INPUT ;C$
10494  COLOR@ (15,6)-(31,6),5:IF C$<>" " THEN PT=VAL(C$)
10495  COLOR@ (15,7)-(31,7),7:LOCATE 47,7 :INPUT ;C$
10496  COLOR@ (15,7)-(31,7),5:IF C$<>" " THEN PE=VAL(C$)
10497  COLOR@ (15,8)-(31,8),7:LOCATE 47,8 :INPUT ;C$

```



```

10498 COLOR@ (15,8)-(31,8),5:IF C$<>" THEN PD=VAL(C$)
10499 COLOR@ (15,9)-(31,9),7:LOCATE 47,9 :INPUT ;C$
10500 IF C$<>"Y" AND C$<>"N" AND C$<>" THEN GOTO 10499
10501 COLOR@ (15,9)-(31,9),5:IF C$<>" THEN USE$=C$
10502 IF USE$ ="Y" THEN DIV =10
10503 IF USE$ ="N" THEN DIV = 1
10504 LOCATE 15,10:PRINT "ARE YOU SURE ???(Y/N)"
10505 LOCATE 47,10
10506 IF C$="Y" THEN GOTO 10510 ELSE GOTO *INITSET
10507 '
10508 '
10510 LOCATE 15,10
      :PRINT "OPEN THE SHUTTER AND HIT ANY KEY !!!";
      :C$=INPUT$(1)
10520 '
10530 '----- DISPLAY DURING MEASUREMENT -----
10540 CONSOLE 8,25,0,1:CLS 1 :COLOR 5
10550 LOCATE 15,10 :PRINT "VOLTAGE      (V)"
10560 LOCATE 45,10 :PRINT "CURRENT (mA/cm^2)"
10570 K=0
10580 RETURN
10590 '
10600 '----- DEVICE INITIAL SET -----
10610 *DEVICESET
10620 ISET IFC :ISET REN :CMD DELIM=0
10630 '----- 4140B SET (I) -----
10640 FC=2 ' function          (I-V)
10650 RA=1 ' I-range          (AUTO)
10660 HC=12' limit auto range (1E-12A)
10670 IC=3 ' integration time (LONG)
10680 JC=1 ' filter          (ON)
10690 TC=1 ' I-trigger       (INTERNAL)
10700 AC=3 ' Va-mode         (SINGLE STAIRCASE)
10710 BC=2 ' vb-mode        (OFF)
10720 LC=3 ' Va I-limit     (1E-2A)
10730 WC=1 ' sweep control  (auto start)
10740 '-----
10750 F$="F"+STR$(FC)
10760 RA$="RA"+STR$(RA)
10770 H$="H"+STR$(HC)
10780 I$="I"+STR$(IC)
10790 J$="J"+STR$(JC)
10800 T$="T"+STR$(TC)
10810 A$="A"+STR$(AC)
10820 B$="B"+STR$(BC)
10830 L$="L"+STR$(LC)
10831 PH=PD
10832 PS$="PS"+STR$(PS)
10833 PT$="PT"+STR$(PT)
10834 PE$="PE"+STR$(PE)
10835 PH$="PH"+STR$(PH)
10836 PD$="PD"+STR$(PD)
10840 W$="W"+STR$(WC):PC1$=F$+RA$+H$+I$+J$+T$+A$+B$+L$
10841 PC2$=PS$+" "+PT$+" "+PE$+" "+PH$+" "+PD$
10850 PRINT@ 16;PC1$ : PRINT@ 16;PC2$ : PRINT@ 16;W$ : RETURN
10860 '

```

```

10870 '----- DATA READ ROUTINE -----
10880 *JREAD
10890 INPUT@ 16; A$,B$
10891 '
10900 V(I)=VAL(MID$(B$,2,7))
10910 J(I)=VAL(MID$(A$,3,10))*-1000*DIV/AREA
10915 STS$=MID$(A$,1,1)
10920 RETURN
10930 '
11050 '----- DATA CALUCULATION -----
11060 *PARACALC
11070 VOC=(V(ITOT)+V(ITOT-1))/2
11080 JSC=J(1)
11090 '
11100 WMAX=0:MAXNUM=1
11110 FOR I=1 TO ITOT
11120 W=J(I)*V(I)
11130 IF WMAX<W THEN WMAX=W :MAXNUM=I
11140 NEXT I
11150 '
11160 FF=WMAX*100/(JSC*VOC)
11170 CONV=WMAX*100/PIN
11180 '
11190 RETURN
11200 '
11210 '----- DEVICE RESET -----
11220 *DVCRESET
11230 PRINT@ 16;"w7"
11240 IRESET REN
11250 RETURN
11260 '
11290 '----- DATA DISPLAY ROUTINE -----
11300 *WDATA
11310 COLOR 7
11320 IF I MOD 10=1 THEN K=K+10 :CONSOLE 11,10,0,1 :CLS 1
11330 LOCATE 20,20+(I-K) :PRINT V(I);
11340 LOCATE 48,20+(I-K) :PRINT J(I);
11350 RETURN
11360 '
20000 '----- DATA TREATMENT -----
20010 '
20020 *DATATRE:CLS 3
20030 COLOR 6:LOCATE 27,0
20040 PRINT "*** DATA TREATMENT ***"
20050 COLOR 5:LOCATE 31,2:PRINT "MODE SELECTION"
20060 LOCATE 27,4 :PRINT "1 LOAD DATA"
20070 LOCATE 27,6 :PRINT "2 GO TO TASK SELECTION"
20080 LOCATE 27,10:PRINT "PRESS YOUR CHOICE NUMBER"
20090 COLOR@ (27,4)-(27,10),2:COLOR@ (27,10)-(51,10),2
20100 *LOOP100:C$=INKEY$
20110 IF C$="" THEN *LOOP100
20120 IF C$="1" THEN *CAL
20130 IF C$="2" THEN RETURN
20140 GOTO *DATATRE
20150 '
20170 '----- DATA CALUCULATION -----

```



```

20180 *CAL
20190  CONSOLE 0,25,0,1 :CLS 1 :COLOR 7
20200  GOSUB *READFILE
20210  GOSUB *LDDISP
20220  GOSUB *DISPLY
20230  RETURN
20240  '
20260  '----- GRAPHIC DISPLAY ON CRT -----
20270  *DISPLY
20280  GOSUB *MINMAX: GOSUB *XYDTMN
20290  GOSUB *COMGRA :GOSUB *CRT :GOSUB *HC :GOTO *DISPLY
20300  '
20310  '----- MINMUM AND MAXMUN -----
20320  *MINMAX :PMAX=-1E+10 :PMIN=0          :TMAX=-1E+10 :TMIN=0
20330  FOR I=1 TO ITOT
20340  'IF T(I)<TMIN THEN TMIN=T(I)
20350  'IF P(I)<PMIN THEN PMIN=P(I)
20360  IF T(I)>TMAX THEN TMAX=T(I)
20370  IF P(I)>PMAX THEN PMAX=P(I)
20380  NEXT I
20390  RETURN
20400  '
20410  '----- XMIN,XMAX,YMIN,YMAX DETERMINATION -----
20420  *XYDTMN
20430  CONSOLE 0,25,0,1 :CLS 1 :COLOR 7
20440  PRINT "XMIN=";TMIN
20450  PRINT "XMAX=";TMAX
20460  PRINT "YMIN=";PMIN
20470  PRINT "YMAX=";PMAX :COLOR 6
20480  LOCATE 18,2 :PRINT "PLOTING TASK AND SELECTION" :COLOR 5
20490  LOCATE 21,4 :PRINT "No.=0 X-axis - Y-axis" :COLOR 0
20500  LOCATE 19,6 :PRINT "RETURN Plotting "
20510  LOCATE 21,8 :PRINT "1  New Ref. Loading"
20520  LOCATE 21,9 :PRINT "2  Store this DATA"
20530  LOCATE 21,10:PRINT "3  Next Measurement"
20540  COLOR@ (21,8)-(21,10),2 : COLOR@ (19,6)-(24,6),2
20545  COLOR@ (25,6)-(45,10),5 : COLOR 5
20550  LOCATE 11,12
      :INPUT "No. to choose ITEM ;Press RETURN";IOP0
20560  IF (IOP0<>1) AND (IOP0<>2) AND (IOP0<>3) THEN IOP0=0
20580  IF IOP0=1 THEN *DATATRE
20583  IF IOP0=2 THEN GOSUB *STOREDATA :RETURN *MEASURE
20586  IF IOP0=3 THEN RETURN *MEASURE
20590  IOP11=IOP0 MOD 2
20600  IOP12=IOP0 ¥ 2
20610  *JUMP11 :TMIN=0 :TMAX=1 :PMIN=0 :PMAX=20
20620  LOCATE 21,14:PRINT "X-Axis From ( ? )";TAB(41);TMIN 'XMIN
20630  LOCATE 21,15:PRINT "      To ( ? )";TAB(41);TMAX 'XMAX
20640  LOCATE 21,16:PRINT "Y-Axis From ( ? )";TAB(41);PMIN 'YMIN
20650  LOCATE 21,17:PRINT "      To ( ? )";TAB(41);PMAX 'YMAX
20660  COLOR@ (21,14)-(55,14),7 :LOCATE 55,14 :PRINT SPC(20)
20670  LOCATE 55,14 :INPUT "";XMIN$
20680  IF XMIN$="" THEN XMIN=TMIN ELSE XMIN=VAL(XMIN$)
20690  LOCATE 39,15 :PRINT SPC(1) :COLOR@ (21,14)-(79,14),5
20700  COLOR@ (21,15)-(55,15),7 :LOCATE 55,15 :PRINT SPC(20)
20710  LOCATE 55,15 :INPUT "";XMAX$

```

```

20720 IF XMAX$="" THEN XMAX=TMAX ELSE XMAX=VAL(XMAX$)
20730 LOCATE 39,15 :PRINT SPC(1) :COLOR@ (21,15)-(79,15),5
20740 IF XMIN>=XMAX THEN *JUMP11
20750 IF (IOP11=1) AND ((XMIN<=0) OR (XMAX<=0)) THEN *JUMP11
20760 *JUMP12 :COLOR@ (21,16)-(55,16),7 :LOCATE 55,16 :PRINT SPC(20)
20770 LOCATE 55,16 :INPUT "";YMIN$
20780 IF YMIN$="" THEN YMIN=PMIN ELSE YMIN=VAL(YMIN$)
20790 LOCATE 39,16 :PRINT SPC(1) :COLOR@ (21,16)-(79,16),5
20800 COLOR@ (21,17)-(55,17),7 :LOCATE 55,17 :PRINT SPC(20)
20810 LOCATE 55,17 :INPUT "";YMAX$
20820 IF YMAX$="" THEN YMAX=PMAX ELSE YMAX=VAL(YMAX$)
20830 LOCATE 39,17 :PRINT SPC(1) :COLOR@ (21,17)-(79,17),5
20840 IF YMIN>=YMAX THEN *JUMP12
20850 IF (IOP12=1) AND ((YMIN<=0) OR (YMAX<=0)) THEN *JUMP12
20860 RETURN
20870 '
20880 ' ----- COMMON GRAPHIC ROUTINE ON CRT AND PLOTTER -----
20890 *COMGRA
20900 ZMAX=XMAX
20910 ZMIN=XMIN
20920 IOP1=IOP11
20930 GOSUB *SCALE
20940 IOPX3=IOP3
20950 IOPX4=IOP4
20960 NXT=NZT
20970 XORG2=ZORG2
20980 XT=ZT
20990 ZMAX=YMAX
21000 ZMIN=YMIN
21010 IOP1=IOP12
21020 GOSUB *SCALE
21030 IOPY3=IOP3
21040 IOPY4=IOP4
21050 NYT=NZT
21060 YORG2=ZORG2
21070 YT=ZT
21080 RETURN
21090 '
21100 ' ----- SCALING SUBROUTINE -----
21110 *SCALE
21120 IF IOP1=1 THEN *JUMP13
21130 LZ=INT(LOG(ZMAX)/C1)
21140 ZT=10^LZ
21150 IZORG1=INT(ZMIN/ZT)
21160 ZORG1=ZT*IZORG1
21170 IZORG2=INT(ZMAX/ZT)
21180 IF IZORG2<>ZMAX/ZT THEN IZORG2=IZORG2+1
21190 ZORG2=ZT*IZORG2
21200 GOTO *JUMP14
21210 *JUMP13:OZ=LOG(ZMAX)/C1
21220 IZORG2=INT(OZ)
21230 IF IZORG2<>OZ THEN IZORG2=IZORG2+1
21240 IZORG1=INT(LOG(ZMIN)/C1)
21250 ZORG2=IZORG2
21260 ZORG1=IZORG1
21270 ZT=1

```



```

21280 *JUMP14
21290 IOP2=1
21300 NZT=IZORG2-IZORG1
21310 IF (IOP1=0) AND (NZT=1) THEN ZT=ZT*.1:NZT=10
21320 IF (IOP1=0) AND (NZT=2) THEN IOP2=2
21330 NZT=NZT*IOP2
21340 ZT=ZT/IOP2
21350 IOP3=1
21360 IF (IOP1=0) AND ((NZT>5) OR (IOP2=2)) THEN IOP3=2
21370 IOP4=1
21380 IF (IOP1=0) AND (IZORG2 MOD 2<>0) AND
      (IZORG2-IZORG1<>1) AND (IOP3=2) AND (IOP2=1) THEN IOP4=2
21390 RETURN
21400 '
21410 ' ----- PLOT ON CRT -----
21420 *CRT
21430 IPXORG1=100
21440 IPXORG2=500
21450 IPYORG1=340
21460 IPYORG2=40
21470 CONSOLE 0,25,0,1 :SCREEN 3,0,0,1 :COLOR 7,0,0,5
21480 IPXTIT=(IPXORG1+IPXORG2)*.5-LEN(XTIT$)*7
21490 IPYTIT=(IPYORG1+IPYORG2)*.5-LEN(YTIT$)*10
21500 PXT=(IPXORG2-IPXORG1)/NXT
21510 FOR I=1 TO NXT+1
21520 IPXLAB(I)=IPXORG2-PXT*(I-1)
21530 XLAB$(I)=STR$(XORG2-XT*(I-1))
21540 'IF IOP11=1 THEN XLAB$(I)="1E"+XLAB$(I)
21550 NEXT I
21560 PYT=(IPYORG1-IPYORG2)/NYT
21570 FOR I=1 TO NYT+1
21580 IPYLAB(I)=IPYORG2+PYT*(I-1)
21590 YY=YORG2-YT*(I-1)
21600 IF ABS(YY)<.0001 THEN YY=0
21610 YLAB$(I)=STR$(YY)
21620 NEXT I
21630 A=PXT/XT:B=PYT/YT
21640 FOR I=1 TO ITOT
21650 IF IOP11=1 THEN *JUMP15
21660 IX(I)=IPXORG2+(T(I)-XORG2)*A :GOTO *JUMP16
21670 *JUMP15 :IF T(I)<=0 THEN IX(I)=IPXORG1 :GOTO *JUMP16
21680 IX(I)=IPXORG2+(LOG(T(I))/C1-XORG2)*A
21690 *JUMP16 :IF IX(I)<IPXORG1 THEN IX(I)=IPXORG1
21700           IF IX(I)>IPXORG2 THEN IX(I)=IPXORG2
21710 IF IOP12=1 THEN *JUMP17
21720 IY(I)=IPYORG2-(P(I)-YORG2)*B :GOTO *JUMP18
21730 *JUMP17 :IF P(I)<=0 THEN IY(I)=IPYORG1 :GOTO *JUMP18
21740 IY(I)=IPYORG2-(LOG(P(I))/C1-YORG2)*B
21750 *JUMP18 :IF IY(I)>IPYORG1 THEN IY(I)=IPYORG1
21760           IF IY(I)<IPYORG2 THEN IY(I)=IPYORG2
21770 NEXT I :BBB=B
21780 SCREEN 3,0,0,0 :CLS 3
21790 POINT (IPXORG2,IPYORG1)
21800 FOR I=2 TO NXT+1
21810 LINE -(IPXLAB(I),IPYORG1) :LINE -STEP(0,-8) :POINT STEP(0,8)
21820 NEXT I

```

```

21830 FOR I=NYT TO 1 STEP -1
21840 LINE -(IPXORG1,IPYLAB(I)) :LINE -STEP(4,0) :POINT STEP(-4,0)
21850 NEXT I
21860 FOR I=NXT TO 1 STEP -1
21870 LINE -(IPXLAB(I),IPYORG2) :LINE -STEP(0,8) :POINT STEP(0,-8)
21880 NEXT I
21890 FOR I=2 TO NYT+1
21900 LINE -(IPXORG2,IPYLAB(I)) :LINE -STEP(-4,0) :POINT STEP(4,0)
21910 NEXT I
21920 FOR I=IOPX4 TO NXT+1 STEP IOPX3
21930 A=LEN(XLAB$(I))
21940 IF IOP11=1 THEN *JUMP19
21950 PXL=8*A*.5
21960 FOR J=1 TO A
21970 A2$=MID$(XLAB$(I),J,1)
21980 PUT (IPXLAB(I)-PXL+8*(J-1),IPYORG1+6),KANJI(ASC(A2$)),PSET
21990 NEXT J
22000 GOTO *JUMP20
22010 *JUMP19 :PUT(IPXLAB(I)-15,IPYORG1+13),KANJI(49),PSET
22020 PUT (IPXLAB(I)-7,IPYORG1+13),KANJI(48),PSET
22030 FOR J=1 TO A
22040 A2$=MID$(XLAB$(I),J,1)
22050 PUT (IPXLAB(I)+7*(J-1),IPYORG1+5),KANJI(ASC(A2$)+256),PSET
22060 NEXT J
22070 *JUMP20 :NEXT I
22080 AA=0
22090 FOR I=IOFY4 TO NYT+1 STEP IOFY3
22100 A=LEN(YLAB$(I))
22110 IF A>AA THEN AA=A
22120 NEXT I
22130 FOR I=IOFY4 TO NYT+1 STEP IOFY3
22140 A=LEN(YLAB$(I))
22150 IF IOP12=1 THEN *JUMP21
22160 PYL=8*A+5
22170 FOR J=1 TO A
22180 A2$=MID$(YLAB$(I),J,1)
22190 PUT (IPXORG1-PYL+8*(J-1),IPYLAB(I)-7),KANJI(ASC(A2$)),PSET
22200 NEXT J
22210 GOTO *JUMP22
22220 *JUMP21 :PUT (IPXORG1-8*AA-15,IPYLAB(I)),KANJI(49),PSET
22230 PUT (IPXORG1-8*AA-7,IPYLAB(I)),KANJI(48),PSET
22240 FOR J=1 TO A
22250 A2$=MID$(YLAB$(I),J,1)
22260 PUT (IPXORG1+8*(-AA+J-1),IPYLAB(I)-8),KANJI(ASC(A2$)+256),PSET
22270 NEXT J
22280 *JUMP22 :NEXT I
22290 A=LEN(XTIT$)
22300 B=IPYORG1
22310 IF IOP11=1 THEN B=B+38 ELSE B=B+30
22320 FOR I=1 TO A
22330 A2$=MID$(XTIT$,I,1)
22340 PUT (IPXTIT+14*(I-1),B),KANJI(ASC(A2$)),PSET,5,0
22350 NEXT I
22360 A=LEN(YTIT$)
22370 B=IPXORG1-10-AA*8
22380 IF IOP12=1 THEN B=B-28

```



```

22390 FOR I=1 TO A
22400 A2$=MID$(YTIT$,I,1)
22410 PUT (B,IPYTIT+17*(I-1)),KANJI(ASC(A2$)),PSET,5,0
22420 NEXT I
22430 FOR I=1 TO ITOT-1
22440 LINE(IX(I),IY(I))-(IX(I+1),IY(I+1)),2
22450 NEXT I
22460 LINE(IPXORG1,IPYORG2+BBB*YORG2)-(IPXORG2,IPYORG2+BBB*YORG2),5
22470 CIRCLE (IX(MAXNUM),IY(MAXNUM)),4,4
22480 SCREEN ,,,1
22490 '
22500 LOCATE 45,3
      :PRINT USING "SAMPLE NAME = & &";SAMPLE$
22510 LOCATE 45,4
      :PRINT USING "DATE = & &";DATTE$
22520 LOCATE 45,5
      :PRINT USING "Area (cm^2)= ##.###";AREA
22530 LOCATE 45,6
      :PRINT USING "Pin (mW/cm^2)= ###";PIN
22540 LOCATE 45,7
      :PRINT USING "Voc (V)= #.#####";VOC
22550 LOCATE 45,8
      :PRINT USING "Jsc (mA/cm^2)= ##.#####";JSC
22560 LOCATE 45,9
      :PRINT USING "Vmax (V)= #.#####";V(MAXNUM)
22570 LOCATE 45,10
      :PRINT USING "Jmax (mA/cm^2)= ##.#####";J(MAXNUM)
22580 LOCATE 45,11
      :PRINT USING "Fill Factor(%)= ##.#####";FF
22590 LOCATE 45,12
      :PRINT USING "Conv.Effi (%)= ##.#####";CONV
22600 '
22610 RETURN
22620 '
22630 ' ----- PLOT ON PLOTTER -----
22640 *PLOTTER
22650 ISET IFC :ISET REN
22655 IPXMAX=3000 :IPYMAX=3000
22660 IPXORG1=200 :IPXORG2=200+1800 :IPYORG1=220 :IPYORG2=220+1200
22670 IPXD=200+1300 :IPYD=220+1150 :IST2=40 :IQT2=30
22680 IPXCOM=200+2330 :IPYCOM=220+1200 :IPXSAM=200+2400
      :IPYSAM=220+1200
22690 PRINT@ 16;"^260,70,¥-260,-70,Z",IPXMAX,IPYMAX,"M",
22700 PRINT@ 16;IPXORG2,IPYORG1,"J1,L0"
22710 IST=60 :IQT=50 :IS10=50 :IQ10=30
22720 ISEXP=30 :IQEXP=18 :ISL=50 :IQL=30
22730 CLS 1 :COLOR 6 :LOCATE 26,1 :PRINT "FORMAT SELECTION":COLOR 5
22740 LOCATE 21,3 :PRINT "ID ITEM" :COLOR 0
22750 LOCATE 21,5 :PRINT "0 Dot with Axes"
22760 LOCATE 21,7 :PRINT "1 Line with Axes"
22770 LOCATE 21,9 :PRINT "2 Dot without Axes"
22780 LOCATE 21,11:PRINT "3 Line without Axes"
22790 LOCATE 21,14:PRINT "4 No Plot"
22800 COLOR@ (21,5)-(21,14),2 :COLOR@ (26,5)-(50,14),5 :COLOR 5
22810 LOCATE 11,16
      :INPUT "Type ID # to choose ITEM ;Press RETURN";IOP5

```

```

22820 IOP51=IOP5 MOD 2 :IOP52=IOP5 ¥ 2
22830 IF IOP5>4 THEN *PLOTTER ELSE IF IOP5=4 THEN RETURN ELSE
      IF IOP51=1 THEN *JUMP23
22840 CONSOLE 3,22,0,1 :CLS 1 :CONSOLE 0,25,0,1 :COLOR 0
22850 LOCATE 21,4 :PRINT "0      DOT"
22860 LOCATE 21,6 :PRINT "1      CIRCLE"
22870 LOCATE 21,8 :PRINT "2      CROSS"
22880 LOCATE 21,10:PRINT "3      TRIANGLE"
22890 LOCATE 21,12:PRINT "4      SCARE"
22900 COLOR@ (21,4)-(21,12),2 :COLOR@ (26,4)-(40,12),5 :COLOR 5
22910 LOCATE 11,14
      :INPUT "Type ID # to choose ITEM ;Press RETURN";IDOT
22920 IF (IDOT=3) OR (IDOT=4) THEN IDOT=7-IDOT
22930 IF (IDOT=0) OR (IDOT=1) THEN IDOT=IDOT*4+1
22940 IF IDOT=2 THEN IDOT=7
22950 GOTO *JUMP24
22960 *JUMP23 :CONSOLE 3,22,0,1 :CLS 1 :CONSOLE 0,25,0,1 :COLOR 0
22970 LOCATE 21,4 :PRINT "0      -----"
22980 LOCATE 21,6 :PRINT "1      - - - -"
22990 LOCATE 21,8 :PRINT "2      -%-%-%"
23000 LOCATE 21,10:PRINT "3      -%-%-%"
23010 COLOR@ (21,4)-(21,10),2 :COLOR@ (26,4)-(40,10),5 :COLOR 5
23020 LOCATE 11,12
      :INPUT "Type ID # to choose ITEM ;Press RETURN";LINNE
23030 IF LINNE=0 THEN LINNE=-1
23040 LINNE=LINNE*2+2
23050 *JUMP24 :IPEN=1 'OCATE 21,16 :INPUT "Pen # ";IPEN
23055 LOCATE 21,18
      :INPUT "INFORMATION (NEED = 1) or (NOT = ELSE)";INFO
23060 IPXTIT=(IPXORG1+IPXORG2)*.5-((LEN(XTIT$)-1)*IQT+IST*.57)*.5
23070 IPYTIT=(IPYORG1+IPYORG2)*.5-((LEN(YTIT$)-1)*IQT+IST*.57)*.5
23080 PXT=(IPXORG2-IPXORG1)/NXT
23090 FOR I=1 TO NXT+1
23100 IPXLAB(I)=IPXORG2-PXT*(I-1)
23110 XLAB(I)=XORG2-XT*(I-1)
23120 NEXT I
23130 PYT=(IPYORG2-IPYORG1)/NYT
23140 FOR I=1 TO NYT+1
23150 IPYLAB(I)=IPYORG2-PYT*(I-1)
23160 YLAB(I)=YORG2-YT*(I-1)
23170 NEXT I
23180 A=PXT/XT :B=PYT/YT
23190 FOR I=1 TO ITOT
23200 IF IOP11=1 THEN *JUMP25
23210 IX(I)=IPXORG2+(T(I)-XORG2)*A
23220 GOTO *JUMP26
23230 *JUMP25 :IF T(I)<=0 THEN IX(I)=IPXORG1 :GOTO *JUMP26
23240 IX(I)=IPXORG2+(LOG(T(I))/C1-XORG2)*A
23250 *JUMP26 :IF IOP12=1 THEN *JUMP27
23260 IY(I)=IPYORG2+(P(I)-YORG2)*B
23270 GOTO *JUMP28
23280 *JUMP27 :IF P(I)<=0 THEN IY(I)=IPYORG1 :GOTO *JUMP28
23290 IY(I)=IPYORG2+(LOG(P(I))/C1-YORG2)*B
23300 *JUMP28 :NEXT I
23320 IF IOP52=1 THEN *JUMP61
23330 FOR I=2 TO NXT+1

```



```

23340 PRINT@ 16;"D",IPXLAB(I),IPYORG1,CHR$(3),"E0,20,0,-20",CHR$(3)
23350 NEXT I
23360 FOR I=NYT TO 1 STEP -1
23370 PRINT@ 16;"D",IPXORG1,IPYLAB(I),CHR$(3),"E20,0,-20,0",CHR$(3)
23380 NEXT I
23390 FOR I=NXT TO 1 STEP -1
23400 PRINT@ 16;"D",IPXLAB(I),IPYORG2,CHR$(3),"E0,-20,0,20",CHR$(3)
23410 NEXT I
23420 FOR I=2 TO NYT+1
23430 PRINT@ 16;"D",IPXORG2,IPYLAB(I),CHR$(3),"E-20,0,20,0",CHR$(3)
23440 NEXT I
23450 IA10=IQ10+IS10*.57
23460 FOR I=IOPX4 TO NXT+1 STEP IOPX3
23470 IF IOP11=1 THEN *JUMP29
23480 A=LEN(STR$(XLAB(I)))
23490 PXL=IQL*A*.5
23500 PRINT@ 16;"M",IPXLAB(I)-PXL,IPYORG1+75-ISL-20,
      "S",ISL,"Q",IQL,"P"
23510 PRINT@ 16;XLAB(I)
23520 PRINT@ 16;CHR$(3)
23530 GOTO *JUMP30
23540 *JUMP29 :A=IPYORG1-ISEXP-20
23550 PRINT@ 16;"M",IPXLAB(I)-IA10,A-IS10,
      "S",IS10,"Q",IQ10,"P10"
23560 PRINT@ 16;CHR$(3)
23570 PRINT@ 16;"M",IPXLAB(I),A+IS10,
      "S",ISEXP,"Q",IQEXP,"P"
23580 PRINT@ 16;XLAB(I)
23590 PRINT@ 16;CHR$(3)
23600 *JUMP30 :NEXT I
23610 FOR I=IOPY4 TO NYT+1 STEP IOPY3
23620 A=LEN(STR$(YLAB(I)))
23630 IF A>AA THEN AA=A
23640 NEXT I
23650 FOR I=IOPY4 TO NYT+1 STEP IOPY3
23660 IF IOP12=1 THEN *JUMP31
23670 PYL=IQL*AA
23680 PRINT@ 16;"M",IPXORG1-PYL-20,IPYLAB(I)+75-ISL*.5,
      "S",ISL,"Q",IQL,"P"
23690 PRINT@ 16;YLAB(I)
23700 PRINT@ 16;CHR$(3)
23710 GOTO *JUMP32
23720 *JUMP31 :A=IPXORG1-IQEXP*AA-10
23730 PRINT@ 16;"M",A-IA10,IPYLAB(I)-IS10,
      "S",IS10,"Q",IQ10,"P10"
23740 PRINT@ 16;CHR$(3)
23750 PRINT@ 16;"M",A,IPYLAB(I)+IS10,
      "S",ISEXP,"Q",IQEXP,"P"
23760 PRINT@ 16;YLAB(I)
23770 PRINT@ 16;CHR$(3)
23780 *JUMP32 :NEXT I
23790 A=IPYORG1-IST-70+90
23800 IF IOP11=1 THEN A=A-ISEXP-IS10 ELSE A=A-ISL
23810 PRINT@ 16;"M",IPXTIT,A,"S",IST,"Q",IQT,"P"
23820 PRINT@ 16;XTIT$
23830 PRINT@ 16;CHR$(3)

```

```

23840 A=IPXORG1-IST-85
23850 IF IOP12=1 THEN A=A-AA*IQEXP-IA10 ELSE A=A-AA*IQL
23860 PRINT@ 16;"M",A,IPYTIT,"S",IST,"Q",IQT,"R900,P"
23870 PRINT@ 16;YTIT$
23880 PRINT@ 16;CHR$(3)
23890 PRINT@ 16;"A"
23895 IF INFO<>1 THEN *JUMP61
23900 PRINT@ 16;"M",IPXSAM,IPYSAM,"S",IST,"Q",IQT,"R-900,P"
23910 PRINT@ 16;SAMPLE$
23920 PRINT@ 16;CHR$(3)
23930 PRINT@ 16;"M",IPXCOM,IPYCOM,"P"
23940 PRINT@ 16;COMMEN$
23950 PRINT@ 16;CHR$(3)
23960 PRINT@ 16;"A"
23970 PRINT@ 16;"S",IST2,"Q",IQT2
23980 PRINT@ 16;"M",IPXD,IPYD,"P"
23990 PRINT@ 16;"Date          "+DATE$
24000 PRINT@ 16;CHR$(3)
24010 PRINT@ 16;"M",IPXD,IPYD-(IST2+10),"P"
24020 PRINT@ 16;"Area    (cm^2)" +STR$(AREA)
24030 PRINT@ 16;CHR$(3)
24040 PRINT@ 16;"M",IPXD,IPYD-(IST2+10)*2,"P"
24050 PRINT@ 16;"Pin    (mW/cm^2)" +STR$(PIN)
24060 PRINT@ 16;CHR$(3)
24070 PRINT@ 16;"M",IPXD,IPYD-(IST2+10)*3,"P"
24080 PRINT@ 16;"Voc    (Volts)" +STR$(VOC)
24090 PRINT@ 16;CHR$(3)
24100 PRINT@ 16;"M",IPXD,IPYD-(IST2+10)*4,"P"
24110 PRINT@ 16;"Jsc    (mA/cm^2)" +STR$(JSC)
24120 PRINT@ 16;CHR$(3)
24130 PRINT@ 16;"M",IPXD,IPYD-(IST2+10)*5,"P"
24140 PRINT@ 16;"Vmax    (Volts)" +STR$(V(MAXNUM))
24150 PRINT@ 16;CHR$(3)
24160 PRINT@ 16;"M",IPXD,IPYD-(IST2+10)*6,"P"
24170 PRINT@ 16;"Jmax    (mA/cm^2)" +STR$(J(MAXNUM))
24180 PRINT@ 16;CHR$(3)
24190 PRINT@ 16;"M",IPXD,IPYD-(IST2+10)*7,"P"
24200 PRINT@ 16;"Fill factor(%)" +STR$(FF)
24210 PRINT@ 16;CHR$(3)
24220 PRINT@ 16;"M",IPXD,IPYD-(IST2+10)*8,"P"
24230 PRINT@ 16;"Conv. effi. (%)" +STR$(CONV)
24240 PRINT@ 16;CHR$(3)
24250 PRINT@ 16;"A"
24260 *JUMP61
      :PRINT@ 16;"%",IPXORG1,IPYORG1,"",Z",IPXORG2,IPYORG2
24270 IF IOP5=1 OR IOP5=3 THEN *JUMP33
24280 FOR I=1 TO ITOT
24290 PRINT@ 16;"M",IX(I),IY(I),"",J",IPEN,"",S20,N",IDOT,"",A"
24300 NEXT I
24310 GOTO *JUMP34
24320 *JUMP33
24330 PRINT@ 16;"M",IX(1),IY(1),"",J",IPEN,"",L",LINNE
24340 FOR I=2 TO ITOT
24350 PRINT@ 16;"D",IX(I),IY(I)
24360 NEXT I
24370 PRINT@ 16;CHR$(3),"A"

```



```

24380 *JUMP34
24390 PRINT@ 16;"M",IX(MAXNUM),IY(MAXNUM),
      " ,J",IPEN," ,S40,N",5," ,A"
24400 PRINT@ 16;"¥-260,-70,Z",IPXMAX,IPYMAX
24410 RETURN
24420 '
24430 ' ----- HARD COPY THE GRAPH ON CRT or NOT -----
24440 *HC
24450 LOCATE 14,5 :PRINT "*** Hardcopy ***"
24460 LOCATE 14,6 :PRINT "Figure          ---(1)"
24470 LOCATE 14,7 :PRINT "Figure and Inform.(2)"
24480 LOCATE 14,8 :PRINT "PLOTTER------(3)"
24490 LOCATE 14,9 :PRINT "Not          -----(RETURN)"
24500 LOCATE 14,10:INPUT "";IOP8
24505 IF (IOP8<>1) AND (IOP8<>2) AND (IOP8<>3) THEN RETURN
24510 IF IOP8=3 THEN GOSUB *PLOTTER :RETURN
24520 IF (IOP8=1) OR (IOP8=2) THEN LPRINT CHR$(27);"!":
      :COPY 2 :LPRINT CHR$(27);CHR$(34)
24530 IF IOP8<>2 THEN LPRINT CHR$(12) :RETURN
24540 LPRINT USING "SAMPLE NAME   = &          & ";SAMPLE$;
24550 LPRINT USING "DATE           = &          & ";DATTE$
24560 LPRINT USING "Area      (cm^2)= ##.###          ";AREA;
24570 LPRINT USING "Pin      (mW/cm^2)= ###          ";PIN
24580 LPRINT USING "Voc       (V)= #.####          ";VOC;
24590 LPRINT USING "Jsc      (mA/cm^2)= ##.####          ";JSC
24600 LPRINT USING "Vmax      (V)= #.####          ";V(MAXNUM);
24610 LPRINT USING "Jmax     (mA/cm^2)= ##.####          ";J(MAXNUM)
24620 LPRINT USING "Fill Factor(%)= ##.####          ";FF;
24630 LPRINT USING "Conv.Effi (%)= ##.####          ";CONV
24640 '
24650 LPRINT CHR$(12)
24660 RETURN
24670 '
24680 ' ----- DATA STORAGE -----
24690 *STOREDATA :CLS 3
24700 COLOR 6:LOCATE 10,0
24710 PRINT "*** STORE THE DATA ***"
24720 COLOR 5:LOCATE 40,2 :PRINT "1-----8"
24730 *JUMP106 :LOCATE 10,3:PRINT "DATE          (8 CHARACTERS)"
      :LOCATE 40,3:PRINT DATTE$
24735 COLOR 7 :LOCATE 40,3:LINE INPUT C$
      :IF C$ <> "" THEN DATTE$ = C$
24740 IF LEN(DATTE$)>8 THEN *JUMP106
24750 COLOR 5:LOCATE 40,5 :PRINT "1-----20"
24760 *JUMP107 :LOCATE 10,6:PRINT "SAMPLE NAME(20CHARACTERS)"
      :LOCATE 40,6:PRINT SAMPLE$
24765 COLOR 7 :LOCATE 40,6:LINE INPUT C$
      :IF C$<>"" THEN SAMPLE$=C$
24770 IF LEN(SAMPLE$)>20 THEN *JUMP107
24780 COLOR 5 :LOCATE 40,8 :PRINT "1-----10" : FILE$ = SAMPLE$
24790 *JUMP104 :LOCATE 10,9 :PRINT "FILE NAME (10 CHARACTERS)"
24795 COLOR 7 :LOCATE 40,9 : PRINT FILE$ : 'LINE INPUT FILE$
24800 IF LEN(FILE$)>10 THEN *JUMP104
24810 *JUMP105 :COLOR 5:LOCATE 40,11:PRINT "1-----20"
24820 LOCATE 10,12:PRINT "COMMENT (20 CHARACTERS)" :COLOR 7
24825 LOCATE 40,12:LINE INPUT COMMEN$

```

```

24830 IF LEN(COMMEN$)>20 THEN *JUMP105
24840 FOR I=1 TO ITOT :P(I)=V(I) :P(I+ITOT)=J(I) :NEXT I
24850 GOSUB *MAKEFILE :RETURN *DISPLY
24860 '
24870 '----- LOADED DATA DISPLY -----
24880 *LDDISP:CLS 3 :COLOR 6
24890 LOCATE 20,0 :PRINT "***** LOADED DATA *****"
24900 COLOR 5:LOCATE 20,2 :PRINT "FILE NAME"
      :COLOR 7:LOCATE 41,2 :PRINT FILE$
24910 COLOR 5:LOCATE 20,3 :PRINT "COMMENT"
      :COLOR 7:LOCATE 41,3 :PRINT COMMEN$
24920 COLOR 5:LOCATE 20,4 :PRINT "SAMPLE NAME "
      :COLOR 7:LOCATE 41,4 :PRINT SAMPLE$
24930 COLOR 5:LOCATE 20,5 :PRINT "DATE"
      :COLOR 7:LOCATE 41,5 :PRINT DATTE$
24940 COLOR 5:LOCATE 20,6 :PRINT "CELL AREA (cm^2)"
      :COLOR 7:LOCATE 41,6 :PRINT STR$(AREA)
24950 COLOR 5:LOCATE 20,7 :PRINT "P IN(mW/cm^2)"
      :COLOR 7:LOCATE 41,7 :PRINT STR$(PIN)
24960 *JUMP1000:COLOR 6
      :LOCATE 20,10 :PRINT "1      HARDCOPY & CONTINUE"
24970 LOCATE 20,11 :PRINT "2 NOT HARDCOPY & CONTINUE"
24980 COLOR@ (20,17)-(20,19),2 :COLOR@ (21,20)-(44,20),2
24990 *LOOP1000:C$=INKEY$
25000 IF C$="" THEN *LOOP1000
25010 IF C$="2" OR C$=CHR$(13)THEN *ESCAPE1000
25020 IF C$<>"1" THEN *JUMP1000
25030 CONSOLE 17,20,0,1 :CLS 3 :CONSOLE 0,25,0,1 :COPY
25040 *ESCAPE1000
25050 RETURN
25060 '--- MAKING SEAQUENTIAL FILE ----
25070 *MAKEFILE
25080 OPEN DV$+FILE$ AS #1
25090 WRITE #1,DATTE$;SAMPLE$;COMMEN$
25100 WRITE #1,FC;RA;HC;IC;JC;TC;AC;BC;LC;WC;PS;PT;PE;PH;PD
25110 WRITE #1,AREA;PIN;USE$
25120 WRITE #1,FF;CONV;VOC;JSC;ITOT
25130 FOR I=1 TO ITOT :WRITE #1,V(I);J(I) :NEXT I
25140 CLOSE #1
25150 RETURN
25160 '
25170 '--- READ SEQUENTIAL FILE ----
25180 *READFILE
25190 FILES DV$ :INPUT "FILE NAME = ";FILE$
25200 OPEN DV$+FILE$ AS #1
25210 INPUT #1,DATTE$,SAMPLE$,COMMEN$
25220 INPUT #1,FC,RA,HC,IC,JC,TC,AC,BC,LC,WC,PS,PT,PE,PH,PD
25230 INPUT #1,AREA,PIN,USE$
25240 INPUT #1,FF,CONV,VOC,JSC,ITOT
25250 FOR I=1 TO ITOT :INPUT #1,V(I),J(I) :NEXT I
25260 CLOSE #1
25270 FOR I=1 TO ITOT :T(I)=V(I):P(I)=J(I):NEXT I
25275 GOSUB *PARACALC
25280 RETURN
25290 '

```


ภาคผนวก ง.

โปรแกรมการวัดกระแสตอบสนองของเซลล์แสงอาทิตย์

```

100 '----- PHOTO CURRENT RESPONSE MESUREMENT -----'
110 '      for hp 4140B pA METER / DC VOLTAGE SOURCE      '
120 '-----'
130 '              19/07/1991 by Suwat Sopitpan              '
140 '-----'
150 '
160 '
180 CONSOLE 0,25,0,1 : CLS : OPTION BASE 1
185 DELAY = 9000 : SELECT = 1 : DOG = 4
190 DIM A$[18]
200 DIM EA(200), CURRENT(200)
205 DIM WLENGTH(200) , CONTENT(200)
210 *SHOW
215 CONSOLE 0,25,0,1 : CLS
220 IF ( SELECT = DOG ) GOTO 990
225 GOSUB *MENU
230 *LOOP1
240 LOCATE 15,20 : COLOR 7 : INPUT "Select : ",SELECT
250 ON SELECT GOSUB *PICO, *FSAVE, *EPSON, *BYE
260 GOTO *SHOW
900 '
1000 *DATSET '----- DATA SET -----'
1010 WLI = 400 'initial wavelength > 400 nm <
1020 WLF = 800 'final wavelength > 800 nm <
1030 WLS = 10 'step > 10 nm <
1040 RETURN
1050 '
1060 '
1160 *DEVSET '----- DEVICE SET IEEE CONTROLLER -----'
1170 ISET IFC
1180 ISET REN
1190 CMD DELIM = 0
1200 RETURN
1210 '
1220 *DEVRESET '----- RESET TO LOCAL CONTROL -----'
1230 IRESET REN
1240 RETURN
1250 '
4000 *PICO
4010 CONSOLE 6,24,0,1 : CLS
4020 COLOR 4
4025 LOCATE 20,6 : PRINT "...ELECTRICAL..."
4027 COLOR 6
4030 LOCATE 20,8 : PRINT "1. Zero ampere set"
4040 LOCATE 20,9 : PRINT "2. Start measurement"
4050 LOCATE 20,10 : PRINT "3. Quit to main"

```

```
4100 LOCATE 20,14 : COLOR 7
4110 INPUT "Select : ",SELECT
4120 ON SELECT GOSUB *ZEROAMP, *AMPARE
4130 IF SELECT = 3 THEN *PICOEND
4140 GOTO *PICO
4150 *PICOEND
4160 RETURN
4170 '
4180 '
4190 *AMPARE
4195 CLS : LOCATE 10,9 : COLOR 1
4200 PRINT "-- Electrical measurement --" : COLOR 5
4210 LOCATE 10,11 : INPUT "Start wavelength : ",EINIT
4220 LOCATE 10,13 : INPUT "Stop wavelength : ",B
4230 GOSUB *PICASET
4235 LOCATE 10,15 : COLOR 6 : PRINT "WAVELENGTH           CURRENT"
4237 N = 1 : EA(N) = EINIT
4240 *LOOP3
4250 INPUT "",A$ : INPUT @16; A$ : GOSUB *TOOD
4255 B$ = MID$(A$,3,10)
4257 CURRENT(N) = VAL(B$)
4260 LOCATE 12,17 : COLOR 2 : PRINT EA(N), CURRENT(N)
4270 N = N + 1 : EA(N) = EA(N-1) + 10
4280 IF EA(N) > B THEN *EXIT3
4290 GOTO *LOOP3
4300 *EXIT3
4310 BEEP 1 : FOR C = 1 TO DELAY : NEXT C : BEEP 0
4320 '
4330 MAXELC = N-1
4340 GOSUB *DEVRESET
4350 RETURN
4360 '
4390 '
4400 *TOOD
4410 BEEP 1
4420 FOR TD = 1 TO 500 : NEXT TD
4430 BEEP 0
4440 RETURN
4450 '
5100 *ZEROAMP
5110 ISET IFC : ISET REN : CMD DELIM = 0
5120 '
5130 PRINT @16; "Z"
5140 '
5150 IRESET REN
5160 '
5190 RETURN
5195 '
5300 *PICASET
5310 ISET IFC : ISET REN : CMD DELIM = 0
5320 PRINT @16;"F1RA1HC12JC1A6B2"
5325 RETURN
5330 '
5400 *RESETPICO
5410 IRESET REN
5440 RETURN
```



```

5450 '
6050 '
7000 *FSAVE
7010 COLOR 3 : CLS : LOCATE 20,9
7020 PRINT "---- Save data to disk ----"
7030 COLOR 5
7050 LOCATE 20,12 : INPUT "File name : ",F$
7060 LOCATE 20,13 : INPUT "Comment : ",COMMT$
7065 *SAVEASK
7070 LOCATE 20,16 : INPUT "Sure (1), Not correct (2) or Quit (3) : ",ASK
7080 ON ASK GOTO *GOSAVE, *FSAVE, *EXITSAVE
7090 GOTO *SAVEASK
7100 *GOSAVE
7200 EXTEN$ = ".elc"
7210 FOR I = 1 TO MAXELC
7220 WLENGTH(I) = EA(I) : CONTENT(I) = CURRENT(I)
7230 NEXT I
7310 FILE$ = F$ + EXTEN$
7320 OPEN FILE$ AS #1
7330 WRITE #1 , COMMT$, I-1
7340 FOR N = 1 TO I-1
7350 WRITE #1 , WLENGTH(N), CONTENT(N)
7360 NEXT N
7370 CLOSE #1
7380 *EXITSAVE
7390 RETURN
8050 '
8500 *EPSON
8510 LPRINT "-----"
8512 LPRINT "----- Current response -----"
8520 LPRINT " Wavelength [nm] Current [Amp]"
8530 LPRINT "-----"
8540 FOR N = 1 TO MAXELC
8550 LPRINT EA(N), CURRENT(N)
8560 NEXT N
8570 RETURN
9000 *BYE
9010 SELECT = DOG : CLS
9020 RETURN
9040 '
9280 *MENU : CLS
9290 COLOR 5 : LOCATE 10,3
9300 PRINT "----- PHOTO CURRENT RESPONSE MEASUREMENT -----"
9310 COLOR 2 : LOCATE 10,5
9320 PRINT "-----"
9330 LOCATE 20,8 : COLOR 3
9380 PRINT "1. Electrical measurement"
9390 LOCATE 20,10
9420 PRINT "2. Store data to disk"
9430 LOCATE 20,12
9460 PRINT "3. Print data to printer"
9465 LOCATE 20,14
9467 PRINT "4. Quit"
9470 RETURN

```

ภาคผนวก จ.

การวิเคราะห์ข้อมูลเชิงเศรษฐศาสตร์ ของระบบการผลิตไฟฟ้าจากเซลล์แสงอาทิตย์¹

การวิเคราะห์ในเชิงเศรษฐศาสตร์นี้เป็นการเปรียบเทียบต้นทุนต่อ 1 กิโลวัตต์-ชั่วโมง (หรือเรียกกันทั่วไปว่ายูนิตไฟฟ้า) ระหว่างเซลล์แสงอาทิตย์กับเครื่องยนต์ดีเซลและการขยายสายส่งไฟฟ้าที่ปริมาณการใช้ไฟฟ้าต่อปีต่าง ๆ กัน

ข้อมูลเซลล์แสงอาทิตย์

แผงเซลล์แสงอาทิตย์ในปัจจุบันราคาไม่เกิน 300 บาทต่อวัตต์สูงสุด (Peak watt) ซึ่งบริษัทบางแห่งเสนอขายราคาพิเศษหากซื้อเป็นจำนวนมาก โดยอาจเหลือประมาณ 200 บาทต่อวัตต์สูงสุด ในอนาคตอันใกล้นี้อาจเหลือเพียง 100 บาทต่อวัตต์สูงสุด และคาดว่าหลังจากปี ค.ศ. 2000 ราคาอาจลดลงเหลือเพียง 20 บาทต่อวัตต์สูงสุด

ในการวิเคราะห์นี้จึงใช้ราคา 300 ถึง 20 บาทต่อวัตต์สูงสุด เป็นเกณฑ์และบวกค่าติดตั้งชุดควบคุม รวมทั้งแบตเตอรี่/Inverter 50/100%(BOS) ค่าดำเนินการและบำรุงรักษา 2,000 บาทต่อปี โดยมีอายุการใช้งาน 25 ปี (ตารางที่ จ.1 และ จ.2)

ความสามารถในการผลิตไฟฟ้าเฉลี่ยต่อปีต่อวัตต์สูงสุดที่ติดตั้ง จะเปลี่ยนแปลงตั้งแต่ 1,200 - 1,600 กิโลวัตต์-ชั่วโมง ขึ้นอยู่กับความเข้ม และความสม่ำเสมอของแสงอาทิตย์ สำหรับอัตราดอกเบี้ย หากพิจารณาในอัตราเชิงพาณิชย์ทั่ว ๆ ไป จะคิดที่ 10% หากมีการสนับสนุนช่วยเหลือเงินกู้จากรัฐบาลหรือต่างประเทศแล้ว อาจคิดในอัตราดอกเบี้ยต่ำกว่า หรือไม่คิดเลย

¹ กองวิศวกรรมพลังงานความร้อน.

"การวิเคราะห์ข้อมูลเชิงเศรษฐศาสตร์ ของระบบการผลิตไฟฟ้าจากเซลล์แสงอาทิตย์."

นนทบุรี: การไฟฟ้าฝ่ายผลิตแห่งประเทศไทย, มกราคม 2532. (อัดสำเนา)

ข้อมูลเครื่องยนต์ดีเซล

การวิเคราะห์เปรียบเทียบเกี่ยวกับเครื่องยนต์ดีเซล หากพิจารณาที่เครื่องยนต์ดีเซล ที่มีความมั่นคงในการผลิตไฟฟ้าใกล้เคียงกับเซลล์แสงอาทิตย์แล้ว ควรเป็นเครื่องยนต์ที่มีคุณภาพดี และถ้าจะให้ใกล้เคียงกันมากขึ้น ควรใช้เครื่องยนต์สองเครื่องสลับกันใช้งาน และสำรองไว้ หากอีกเครื่องหนึ่งเกิดการชำรุด ตัวอย่างการใช้งานในลักษณะนี้คือ สถานีทวนสัญญาณวิทยุที่อยู่นบยอดเขาในป่าลึก

ดังนั้นในการเปรียบเทียบ จึงใช้เครื่องยนต์ดีเซล ขนาด 5 กิโลวัตต์ ที่มีคุณภาพดี (ตารางที่ จ.3) และเปรียบเทียบในกรณีใช้ 2 เครื่องด้วย (ตารางที่ จ.4) ราคาของเครื่องยนต์ขนาด 5 กิโลวัตต์นี้ รวมค่าติดตั้งประมาณ 190,000 บาทต่อเครื่อง ค่าดำเนินการและบำรุงรักษา 36,000 บาทต่อปี อายุการใช้งาน 5 ปี ราคาน้ำมันดีเซล 8 บาทต่อลิตร และหากอยู่ในบริเวณที่กันดารมาก ราคาน้ำมันดีเซลอาจแพงกว่านี้

ข้อมูลสายส่งไฟฟ้า

ราคาของการขยาสายส่งไฟฟ้า รวมทั้งหม้อแปลงไฟฟ้า จะเปลี่ยนแปลงได้ ตั้งแต่ 120,000 - 500,000 บาทต่อกิโลเมตร ขึ้นอยู่กับลักษณะภูมิประเทศ และขนาดของสายไฟฟ้า ในการวิเคราะห์นี้ จึงใช้ค่า 120,000 บาทต่อกิโลเมตร (ตารางที่ จ.5) และ 500,000 บาทต่อกิโลเมตร (ตารางที่ จ.6) ระยะทางตั้งแต่ 5, 10, 20, 30, 40 และ 50 กิโลเมตร อายุการใช้งาน 30 ปี ค่าบำรุงรักษา 2% ของต้นทุน อัตราดอกเบี้ย 10% และพิจารณาถึงการสูญเสียพลังงานไฟฟ้าประมาณ 10% ในสายส่ง รวมทั้งต้นทุนของกระแสไฟฟ้า 1.20 - 1.50 บาท

ตัวอย่างการวิเคราะห์เชิงเศรษฐศาสตร์

ผลการวิเคราะห์เปรียบเทียบราคา ของไฟฟ้าระหว่างเซลล์แสงอาทิตย์กับเครื่องยนต์ดีเซล แสดงในรูปที่ จ.1, จ.2, จ.3 และ จ.4 และระหว่างเซลล์แสงอาทิตย์กับการขยาสายส่งไฟฟ้า แสดงในรูปที่ จ.5 และ จ.6

ในกรณีรูปที่ จ.4 เป็นการเปรียบเทียบราคาของไฟฟ้า ระหว่างเซลล์แสงอาทิตย์กับ เครื่องยนต์ดีเซล 2 เครื่อง ซึ่งเป็นการรวมรูปที่ จ.1, จ.2 และ จ.3 เข้าด้วยกัน รูปที่ จ.4 จะมีแกนบอกค่าในแนวนอน 2 แกน คือแกนบนและแกนล่าง โดยที่แกนบน บอกค่าเงินลงทุน ทั้งหมดของเซลล์แสงอาทิตย์ต่อวัตต์ (สูงสุด) และแกนล่างบอกค่า Capacity Factor (ตัวย่อ CF) ของเครื่องยนต์ดีเซล เส้นกราฟเส้นตรง 6 เส้น ที่เป็นเซลล์แสงอาทิตย์ใช้กับ แกนบน และเส้นกราฟโค้ง 3 เส้น ซึ่งเป็นของเครื่องยนต์ดีเซล ใช้กับแกนล่าง สำหรับค่า CF นี้ บ่งบอกถึงเปอร์เซ็นต์ หรือสัดส่วนของพลังงานไฟฟ้าที่ผลิต เทียบกับพลังงานที่ผลิตได้สูงสุด ตลอดทั้งปี คือ $5 \times 8,760 = 43,800$ กิโลวัตต์-ชั่วโมง ค่า 8,760 คือจำนวนชั่วโมง ใน 1 ปี ถ้าเครื่องยนต์นี้มีค่า CF = 15% แสดงว่าผลิตกระแสไฟฟ้า ปีละ $0.15 \times 43,800 = 6,570$ กิโลวัตต์-ชั่วโมง

ผลจากการวิเคราะห์ที่เปรียบเทียบราคาต้นทุนของกระแสไฟฟ้าโดยทั่ว ๆ ไป อาจสรุป ได้ว่า ถ้าต้องการใช้ไฟฟ้าปริมาณน้อย ๆ ต่อปี แนวโน้มเซลล์แสงอาทิตย์จะมีความคุ้มค่ามากขึ้น

ตัวอย่างเช่น หากแผงเซลล์แสงอาทิตย์ราคา 200 บาทต่อวัตต์สูงสุด ผลิตไฟฟ้าได้ 1,500 กิโลวัตต์-ชั่วโมงต่อกิโลวัตต์สูงสุดต่อปี อัตราดอกเบี้ย = 5% และความต้องการ ไฟฟ้าน้อยกว่า 8,000 กิโลวัตต์-ชั่วโมงต่อปี การผลิตไฟฟ้าจากเซลล์แสงอาทิตย์จะถูกกว่าผลิต จากเครื่องยนต์ดีเซลเครื่องเดียว (จุด A รูปที่ จ.2)

ในกรณีรูปที่ จ.4 จะมีลักษณะเดียวกับที่อธิบายมาข้างต้น กล่าวคือในกรณีเงินลงทุน 300 บาทต่อวัตต์ (สูงสุด) (ราคาเซลล์แสงอาทิตย์ 200 บาท และ BOS อีก 100 บาท) ผลิต ไฟฟ้าได้ปีละ 1,400 กิโลวัตต์-ชั่วโมงต่อวัตต์ (สูงสุด) อัตราดอกเบี้ย 10% จากรูปที่ จ.4 (ตามแนวเส้น A B C D) จะได้ค่า Capacity Factor ของเครื่องยนต์ดีเซลที่ 15% หรือ ผลิตไฟฟ้า (หรือความต้องการไฟฟ้า) ปีละ 6,570 กิโลวัตต์-ชั่วโมง แสดงว่าความต้องการ ไฟฟ้าตลอดปี ต่ำกว่า 6,570 กิโลวัตต์-ชั่วโมง การผลิตไฟฟ้าจากเซลล์แสงอาทิตย์จะถูกกว่า จากเครื่องยนต์ดีเซลสองเครื่อง

ในกรณีของการขยายสายส่งไฟฟ้าไปยังหมู่บ้านทั่ว ๆ ไป ที่อยู่ห่างไกล 20 กิโลเมตร และการใช้ไฟฟ้าไม่เกิน 12,000 กิโลวัตต์-ชั่วโมงต่อปี จะแพงกว่าการผลิตไฟฟ้าจาก เซลล์แสงอาทิตย์ ที่ราคาเซลล์แสงอาทิตย์ 200 บาทต่อวัตต์สูงสุด ผลิตไฟฟ้าได้ 1,500 กิโลวัตต์-ชั่วโมงต่อกิโลวัตต์สูงสุดต่อปี อัตราดอกเบี้ย 10% (จุด A รูปที่ จ.5)

ในทำนองเดียวกัน ถ้าหมู่บ้านนั้นถูกรบกวนมาก การขยายสายส่ง จะแพงมากยิ่งขึ้น
ในกรณีเช่นนี้ อยู่ห่างไกลเพียง 5 กิโลเมตร จะแพงกว่าเช่นกัน (จุด A รูปที่ จ.6)

โดยปกติการผลิตไฟฟ้าจากเซลล์แสงอาทิตย์ ไม่มีมลภาวะต่ออากาศ และไม่มีเสียง
รบกวนเลย แต่การผลิตไฟฟ้าจากการเผาไหม้เชื้อเพลิงต่าง ๆ มีมลภาวะต่ออากาศ จน
กระทั่งเป็นปัญหาต่อสภาวะแวดล้อมในขณะนี้ และหากจะเผาไหม้เชื้อเพลิง โดยไม่มีมลภาวะต่อ
อากาศ หรือทำให้เครื่องชนิดเดินเงียบเหมือนเซลล์แสงอาทิตย์แล้ว จะต้องลงทุนเพิ่มขึ้น ซึ่ง
ลักษณะเช่นนี้แนวโน้มของการใช้เซลล์แสงอาทิตย์ก็จะมีศักยภาพสูงขึ้นไปอีก

INVESTMENT COST					
- PLANT COST	[BAHT/kWp]	100000	200000	300000	400000
- BALANCE OF SYSTEM	[BAHT/kWp]	50000	100000	150000	200000
PLANT LIFE					
	[YEAR]	25	25	25	25
ANNUAL INVESTMENT COST					
[BAHT/kWp/Yr.]					
- INTEREST RATE	0%	6000.00	12000.00	18000.00	24000.00
	5%	10642.87	21285.74	31928.61	42571.47
	10%	16525.21	33050.42	49575.63	66100.84
- O & M COST	[BAHT/Yr.]	2000	2000	2000	2000
ELECTRICITY GENERATED					
[kWh/kWp/Yr.]		1200	1200	1200	1200
UNIT COST [BAHT/kWh]					
- INTEREST RATE	0%	6.67	11.67	16.67	21.67
	5%	10.54	19.40	28.27	37.14
	10%	15.44	29.21	42.98	56.75
ELECTRICITY GENERATED					
[kWh/kWp/Yr.]		1500	1500	1500	1500
UNIT COST [BAHT/kWh]					
- INTEREST RATE	0%	5.33	9.33	13.33	17.33
	5%	8.43	15.52	22.62	29.71
	10%	12.35	23.37	34.38	45.40

ตารางที่ ๓.๑ ข้อมูลระบบ Photovoltaic 50% BOS

INVESTMENT COST						
- PLANT COST	[BAHT/kWp]	20000	50000	100000	200000	300000
- BALANCE OF SYSTEM	[BAHT/kWp]	20000	50000	100000	200000	300000
PLANT LIFE						
	[YEAR]		25	25	25	25
ANNUAL INVESTMENT COST						
	[BAHT/kWp/Yr.]					
- INTEREST RATE	0%	1600.00	4000.00	8000.00	16000.00	24000.00
	5%	2838.10	7095.25	14190.49	28380.98	42571.47
	10%	4406.72	11016.81	22033.61	44067.23	66100.84
- O & M COST	[BAHT/Yr.]	2000	2000	2000	2000	2000
ELECTRICITY GENERATED						
	[kWh/kWp/Yr.]	1200	1200	1200	1200	1200
UNIT COST [BAHT/kWh]						
- INTEREST RATE	0%	3.00	5.00	8.33	15.00	21.67
	5%	4.03	7.58	13.49	25.32	37.14
	10%	5.34	10.85	20.03	38.39	56.75
ELECTRICITY GENERATED						
	[kWh/kWp/Yr.]	1500	1500	1500	1500	1500
UNIT COST [BAHT/kWh]						
- INTEREST RATE	0%	2.40	4.00	6.67	12.00	17.33
	5%	3.23	6.06	10.79	20.25	29.71
	10%	4.27	8.68	16.02	30.71	45.40

ตารางที่ จ.2 ข้อมูลระบบ Photovoltaic 100% BOS

INVESTMENT COST								
- PLANT COST	[BAHT]	190000	190000	190000	190000	190000	190000	190000
PLANT LIFE	[YEAR]	5	5	5	5	5	5	5
ANNUAL INVESTMENT COST [BAHT/Yr.]								
- INTEREST RATE	5%	43885.21	43885.21	43885.21	43885.21	43885.21	43885.21	43885.21
	10%	50121.52	50121.52	50121.52	50121.52	50121.52	50121.52	50121.52
- O & M COST	[BAHT/Yr.]	36000	36000	36000	36000	36000	36000	36000
- CAPACITY FACTOR	[%]	3	5	15	20	30	40	50
ELECTRICITY GENERATED [kWh/Yr.]		1314	2190	6570	8760	13140	17520	21900
UNIT COST [BAHT/kWh] (excluded fuel cost)								
- INTEREST RATE	5%	60.80	36.48	12.16	9.12	6.08	4.56	3.65
	10%	65.54	39.32	13.11	9.83	6.55	4.92	3.93
FUEL COST*	[BAHT/kWh]	4.44	4.44	4.44	4.44	4.44	4.44	4.44
UNIT COST [BAHT/kWh] (included fuel cost)								
- INTEREST RATE	5%	65.24	40.92	16.60	13.56	10.52	9.00	8.09
	10%	69.99	43.77	17.55	14.28	11.00	9.36	8.38

* FUEL COST INCLUDES TRANSPORTATION AND STORAGE IN RURAL AREA = 8.00 BAHT/lt.

ตารางที่ จ.3 ข้อมูลระบบ Diesel Generator Power Plant ขนาด 5 kW

INVESTMENT COST								
- PLANT COST	[BAHT]	380000	380000	380000	380000	380000	380000	380000
PLANT LIFE								
	[YEAR]	5	5	5	5	5	5	5
ANNUAL INVESTMENT COST								
	[BAHT/Yr.]							
- INTEREST RATE	5%	87770.42	87770.42	87770.42	87770.42	87770.42	87770.42	87770.42
	10%	100243.04	100243.04	100243.04	100243.04	100243.04	100243.04	100243.04
- O & M COST	[BAHT/Yr.]	36000	36000	36000	36000	36000	36000	36000
- CAPACITY FACTOR	[%]	3	5	15	20	30	40	50
ELECTRICITY GENERATED								
	[kWh/Yr.]	1314	2190	6570	8760	13140	17520	21900
UNIT COST [BAHT/kWh]								
	(excluded fuel cost)							
- INTEREST RATE	5%	94.19	56.52	18.84	14.13	9.42	7.06	5.65
	10%	103.69	62.21	20.74	15.55	10.37	7.78	6.22
FUEL COST* [BAHT/kWh]								
		4.44	4.44	4.44	4.44	4.44	4.44	4.44
UNIT COST [BAHT/kWh]								
	(included fuel cost)							
- INTEREST RATE	5%	98.64	60.96	23.28	18.57	13.86	11.51	10.10
	10%	108.13	66.66	25.18	20.00	14.81	12.22	10.67

* FUEL COST INCLUDES TRANSPORTATION AND STORAGE IN RURAL AREA = 3.00 BAHT/lit.

ตารางที่ 3.4 ข้อมูลระบบ 2 x Diesel Generator Power Plant ขนาด 5 kW

TRANSMISSION LINE COST 120000 BAHT/km LOSS 10 %
 PLANT LIFE 30 YEAR ELECTRICITY COST 1.30 BAHT/kWH
 O & M COST 2 % of INVESTMENT COST


ELEC. GEN. (kWH/Yr.)		5000	8000	10000	12000	15000	18000	20000	25000	30000
CIRCUIT 5 km	INTEREST RATE 0%	8.34	5.70	4.82	4.23	3.65	3.26	3.06	2.71	2.47
	5%	12.53	8.32	6.91	5.98	5.04	4.42	4.11	3.55	3.17
	10%	17.94	11.70	9.62	8.23	6.85	5.92	5.46	4.63	4.07
CIRCUIT 10 km	INTEREST RATE 0%	15.38	10.10	8.34	7.17	5.99	5.21	4.82	4.12	3.65
	5%	23.75	15.33	12.53	10.66	8.78	7.54	6.31	5.73	5.04
	10%	34.58	22.10	17.94	15.17	12.29	10.55	9.62	7.96	6.85
CIRCUIT 20 km	INTEREST RATE 0%	29.46	18.90	15.38	13.03	10.63	9.12	8.34	6.93	5.99
	5%	46.21	29.37	23.75	20.01	16.27	13.77	12.53	10.28	8.78
	10%	67.87	42.91	34.53	29.04	23.49	19.79	17.94	14.61	12.39
CIRCUIT 30 km	INTEREST RATE 0%	43.54	27.70	22.42	18.90	15.38	13.03	11.86	9.75	8.34
	5%	68.66	43.40	34.98	29.37	23.75	20.01	18.14	14.77	12.53
	10%	101.51	63.71	51.23	42.91	34.58	29.04	26.26	21.27	17.94
CIRCUIT 40 km	INTEREST RATE 0%	57.62	36.50	29.46	24.77	20.07	16.94	15.38	12.56	10.69
	5%	91.11	57.43	46.21	38.72	31.24	26.25	23.75	19.26	16.27
	10%	134.44	84.51	67.87	56.77	45.63	38.28	34.58	27.93	23.49
CIRCUIT 50 km	INTEREST RATE 0%	71.70	45.30	36.50	30.63	24.77	20.86	18.30	15.38	13.03
	5%	113.57	71.47	57.43	48.08	38.72	32.42	29.37	23.75	20.01
	10%	167.72	105.32	84.51	70.64	56.77	47.53	42.91	34.58	29.04

ตารางที่ ๓.๕ ข้อมูลระบบสายส่งกำลัง ในระบบ 22 kv. (BAHT/kWH)


TRANSMISSION LINE COST 500000 BAHT/km LOSS 10 %
 PLANT LIFE 30 YEAR ELECTRICITY COST 1.30 BAHT/kwh
 O & M COST 2 % of INVESTMENT COST

ELEC. GEN. (kwh/Yr.)		5000	8000	10000	12000	15000	18000	20000	25000	30000
CIRCUIT 5 km	INTEREST RATE 0%	30.63	19.63	15.97	13.52	11.08	9.45	8.63	7.17	6.19
	5%	48.08	30.54	24.69	20.79	16.93	14.23	12.99	10.68	9.10
	10%	70.64	44.64	35.97	30.19	24.41	20.56	18.64	15.17	12.86
CIRCUIT 10 km	INTEREST RATE 0%	59.97	37.97	30.63	25.74	20.86	17.60	15.97	13.03	11.08
	5%	94.86	59.77	48.08	40.28	32.49	27.29	24.69	20.01	16.99
	10%	139.99	87.98	70.64	59.09	47.53	39.82	35.97	29.04	24.41
CIRCUIT 20 km	INTEREST RATE 0%	118.63	74.63	59.97	50.19	40.41	33.89	30.63	24.77	20.86
	5%	188.41	118.25	94.86	79.26	63.67	53.28	48.08	38.72	32.49
	10%	278.67	174.66	139.99	116.87	93.76	78.35	70.64	56.77	47.53
CIRCUIT 30 km	INTEREST RATE 0%	177.30	111.30	89.30	74.63	59.97	50.19	45.30	36.50	30.63
	5%	281.97	176.72	141.63	118.25	94.86	79.26	71.47	57.43	48.08
	10%	417.36	251.34	209.33	174.66	139.99	116.87	105.32	84.51	70.64
CIRCUIT 40 km	INTEREST RATE 0%	235.97	147.97	118.63	99.08	79.52	66.49	59.97	48.23	40.41
	5%	375.53	235.19	188.41	157.23	126.04	105.23	94.86	76.15	63.67
	10%	556.05	348.02	278.67	232.45	186.22	155.40	139.99	112.25	93.76
CIRCUIT 50 km	INTEREST RATE 0%	294.63	184.63	147.97	123.52	99.08	82.78	74.63	59.97	50.19
	5%	469.08	292.66	235.19	196.21	157.23	131.24	118.25	94.86	79.26
	10%	694.74	434.70	348.02	290.23	232.45	193.92	174.66	139.99	116.87


ตารางที่ 3.6 ข้อมูลระบบสายส่งกำลัง ในระบบ 22 kv. (BAHT/kwh)




รูปที่ ๑.๑ เปรียบเทียบ Unit Cost ระหว่าง PV กับ Diesel
 (PV cost = 300 BAHT/W_p + BOS 50%)




รูปที่ 3.2 เปรียบเทียบ Unit Cost ระหว่าง PV กับ Diesel
 (PV cost = 200 BAHT/W_p + BOS 50%)



รูปที่ ๓.๓ เปรียบเทียบ Unit Cost ระหว่าง PV กับ Diesel
 (PV cost = 100 BAHT/W_p + BOS 50%)




รูปที่ 3.4 เปรียบเทียบ Unit Cost ระหว่าง PV กับ Diesel



1,2 -> PV COST 200 BAHT/Wp + BOS 100% 3,4 -> PV COST 50 BAHT/Wp + BOS 100%

รูปที่ ๓.๕ เปรียบเทียบ Unit Cost ระหว่าง PV กับ สายส่งกำลัง
(ราคาสาย = 120,000 บาท/กม.)



1,2 -> PV COST 200 BAHT/Wp | BOS 100% 3,4 -> PV COST 50 BAHT/Wp + BOS 100%

รูปที่ ๓.๖ เปรียบเทียบ Unit Cost ระหว่าง PV กับ สายส่งกำลัง
(ราคาสาย = 500,000 บาท/กม.)

ประวัติผู้เขียน

นายสุวัฒน์ โสภิตพันธ์ เกิดเมื่อวันที่ 4 ตุลาคม พ.ศ.2505 ที่ อำเภอสมุย จังหวัด
สุราษฎร์ธานี สำเร็จการศึกษาปริญญาวิศวกรรมศาสตรบัณฑิต สาขาวิศวกรรมไฟฟ้า จาก
คณะวิศวกรรมศาสตร์ มหาวิทยาลัยเชียงใหม่ เมื่อปี พ.ศ.2528 และได้เข้าทำงานเป็น
วิศวกรไฟฟ้า ในบริษัทเอกชนเป็นเวลา 3 ปี จากนั้นจึงเข้าศึกษาต่อในระดับปริญญาโท
คณะวิศวกรรมศาสตร์ จุฬาลงกรณ์มหาวิทยาลัย ตั้งแต่ปีการศึกษา 2531